Rules for integrands involving exponentials of inverse hyperbolic tangents

1. $\int u\, e^{n \operatorname{ArcTanh}[a x]}\, dx$

   1. $\int x^m e^{n \operatorname{ArcTanh}[a x]}\, dx$

      1: $\int x^m e^{n \operatorname{ArcTanh}[a x]}\, dx$ when $\frac{n-1}{2} \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n \operatorname{ArcTanh}[z]} = \frac{(1+z)^{\frac{n+1}{2}}}{(1-z)^{\frac{n-1}{2}} \sqrt{1-z^2}}$

Rule: If $\frac{n-1}{2} \in \mathbb{Z}$, then

$$\int x^m e^{n \operatorname{ArcTanh}[a x]}\, dx \longrightarrow \int x^m \frac{(1+a x)^{\frac{n+1}{2}}}{(1-a x)^{\frac{n-1}{2}} \sqrt{1-a^2 x^2}}\, dx$$

Program code:

```
Int[E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  Int[((1+a*x)^((n+1)/2)/((1-a*x)^((n-1)/2)*Sqrt[1-a^2*x^2])),x] /;
FreeQ[a,x] && IntegerQ[(n-1)/2]
```

```
Int[x_^m_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  Int[x^m*((1+a*x)^((n+1)/2)/((1-a*x)^((n-1)/2)*Sqrt[1-a^2*x^2])),x] /;
FreeQ[{a,m},x] && IntegerQ[(n-1)/2]
```

      2: $\int x^m e^{n \operatorname{ArcTanh}[a x]}\, dx$ when $\frac{n-1}{2} \notin \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n \operatorname{ArcTanh}[z]} = \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Rule: If $\frac{n-1}{2} \notin \mathbb{Z}$, then

$$\int x^m e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow \int x^m \frac{(1+a\,x)^{n/2}}{(1-a\,x)^{n/2}}\,dx$$

Program code:

```
Int[E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  Int[(1+a*x)^(n/2)/(1-a*x)^(n/2),x] /;
FreeQ[{a,n},x] && Not[IntegerQ[(n-1)/2]]
```

```
Int[x_^m_.*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  Int[x^m*(1+a*x)^(n/2)/(1-a*x)^(n/2),x] /;
FreeQ[{a,m,n},x] && Not[IntegerQ[(n-1)/2]]
```

2. $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c^2-d^2==0$

1: $\int (e+f\,x)^m\,(c+d\,x)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a\,c+d==0\,\land\,\frac{n-1}{2}\in\mathbb{Z}\,\land\,\left(p\in\mathbb{Z}\,\lor\,p-\frac{n}{2}==0\,\lor\,p-\frac{n}{2}-1==0\right)$

Derivation: Algebraic simplification

Basis: If $a\,c+d==0\,\land\,n\in\mathbb{Z}$, then $(c+d\,x)^n\,e^{n\,\text{ArcTanh}[a\,x]}==c^n\left(1-a^2\,x^2\right)^{n/2}$

Note: The condition $p\in\mathbb{Z}\,\lor\,p-\frac{n}{2}==0\,\lor\,p-\frac{n}{2}-1==0$ should be removed when the rules for integrands of the form $(d+e\,x)^m\,(f+g\,x)^n\,\left(a+b\,x+c\,x^2\right)^p$ when $c\,d^2-b\,d\,e+a\,e^2==0$ are strengthened.

Rule: If $a\,c+d==0\,\land\,\frac{n-1}{2}\in\mathbb{Z}\,\land\,\left(p\in\mathbb{Z}\,\lor\,p-\frac{n}{2}==0\,\lor\,p-\frac{n}{2}-1==0\right)$, then

$$\int (e+f\,x)^m\,(c+d\,x)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow c^n\int (e+f\,x)^m\,(c+d\,x)^{p-n}\left(1-a^2\,x^2\right)^{n/2}\,dx$$

Program code:

```
Int[(c_+d_.*x_)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^n*Int[(c+d*x)^(p-n)*(1-a^2*x^2)^(n/2),x] /;
FreeQ[{a,c,d,p},x] && EqQ[a*c+d,0] && IntegerQ[(n-1)/2] && IntegerQ[2*p]
```

```
Int[(e_.+f_.*x_)^m_.*(c_+d_.*x_)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^n*Int[(e+f*x)^m*(c+d*x)^(p-n)*(1-a^2*x^2)^(n/2),x] /;
FreeQ[{a,c,d,e,f,m,p},x] && EqQ[a*c+d,0] && IntegerQ[(n-1)/2] && (IntegerQ[p] || EqQ[p,n/2] || EqQ[p-n/2-1,0]) && IntegerQ[2*p]
```

2: $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2 c^2 - d^2 == 0 \land (p \in \mathbb{Z} \lor c > 0)$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Note: Since $a^2 c^2 - d^2 == 0$, the factor $\left(1+\frac{d\,x}{c}\right)^p$ will combine with one of the factors $(1+a\,x)^{n/2}$ or $(1-a\,x)^{-n/2}$.

Rule: If $a^2 c^2 - d^2 == 0 \land (p \in \mathbb{Z} \lor c > 0)$, then

$$\int u\,(c+d\,x)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow c^p \int u\left(1+\frac{d\,x}{c}\right)^p \frac{(1+a\,x)^{n/2}}{(1-a\,x)^{n/2}}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^p*Int[u*(1+d*x/c)^p*(1+a*x)^(n/2)/(1-a*x)^(n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c^2-d^2,0] && (IntegerQ[p] || GtQ[c,0])
```

3: $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c^2-d^2==0\,\wedge\,\neg\,(p\in\mathbb{Z}\,\vee\,c>0)$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTanh}[z]}==\frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Note: Since $a^2\,c^2-d^2==0$, the factor $(c+d\,x)^p$ will combine with one of the factors $(1+a\,x)^{n/2}$ or $(1-a\,x)^{-n/2}$ after piecewise constant extraction.

Rule: If $a^2\,c^2-d^2==0\,\wedge\,\neg\,(p\in\mathbb{Z}\,\vee\,c>0)$, then

$$\int u\,(c+d\,x)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx\;\longrightarrow\;\int\frac{u\,(c+d\,x)^p\,(1+a\,x)^{n/2}}{(1-a\,x)^{n/2}}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  Int[u*(c+d*x)^p*(1+a*x)^(n/2)/(1-a*x)^(n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c^2-d^2,0] && Not[IntegerQ[p] || GtQ[c,0]]
```

3. $\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $c^2 - a^2 d^2 == 0$

1: $\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $c^2 - a^2 d^2 == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $p \in \mathbb{Z}$, then $\left(c + \frac{d}{x}\right)^p == \frac{d^p}{x^p} \left(1 + \frac{c x}{d}\right)^p$

Rule: If $c^2 - a^2 d^2 == 0 \land p \in \mathbb{Z}$, then

$$\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx \longrightarrow d^p \int \frac{u}{x^p} \left(1 + \frac{c x}{d}\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$$

Program code:

```
Int[u_.*(c_+d_./x_)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  d^p*Int[u*(1+c*x/d)^p*E^(n*ArcTanh[a*x])/x^p,x] /;
FreeQ[{a,c,d,n},x] && EqQ[c^2-a^2*d^2,0] && IntegerQ[p]
```

2. $\int u\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c^2-a^2 d^2==0\land p\notin\mathbb{Z}$

1. $\int u\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c^2-a^2 d^2==0\land p\notin\mathbb{Z}\land\frac{n}{2}\in\mathbb{Z}$

1: $\int u\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c^2-a^2 d^2==0\land p\notin\mathbb{Z}\land\frac{n}{2}\in\mathbb{Z}\land c>0$

Derivation: Algebraic simplification

Basis: If $\frac{n}{2}\in\mathbb{Z}$, then $e^{n\,\text{ArcTanh}[z]}==\frac{(1+z)^{n/2}}{(1-z)^{n/2}}==(-1)^{n/2}\frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Note: Since $c^2-a^2 d^2==0$, the factor $\left(1+\frac{d}{c\,x}\right)^p$ will combine with the factor $\left(1+\frac{1}{a\,x}\right)^{n/2}$ or $\left(1-\frac{1}{a\,x}\right)^{-n/2}$.

Rule: If $c^2-a^2 d^2==0\land p\notin\mathbb{Z}\land\frac{n}{2}\in\mathbb{Z}\land c>0$, then

$$\int u\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx\longrightarrow(-1)^{n/2}c^p\int u\left(1+\frac{d}{c\,x}\right)^p\frac{\left(1+\frac{1}{a\,x}\right)^{n/2}}{\left(1-\frac{1}{a\,x}\right)^{n/2}}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  (-1)^(n/2)*c^p*Int[u*(1+d/(c*x))^p*(1+1/(a*x))^(n/2)/(1-1/(a*x))^(n/2),x] /;
FreeQ[{a,c,d,p},x] && EqQ[c^2-a^2*d^2,0] && Not[IntegerQ[p]] && IntegerQ[n/2] && GtQ[c,0]
```

2: $\int u\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c^2-a^2 d^2==0\land p\notin\mathbb{Z}\land\frac{n}{2}\in\mathbb{Z}\land c\not>0$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTanh}[z]}==\frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Rule: If $c^2-a^2 d^2==0\land p\notin\mathbb{Z}\land\frac{n}{2}\in\mathbb{Z}\land c\not>0$, then

$$\int u\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow \int u\left(c+\frac{d}{x}\right)^p \frac{(1+a\,x)^{n/2}}{(1-a\,x)^{n/2}}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  Int[u*(c+d/x)^p*(1+a*x)^(n/2)/(1-a*x)^(n/2),x] /;
FreeQ[{a,c,d,p},x] && EqQ[c^2-a^2*d^2,0] && Not[IntegerQ[p]] && IntegerQ[n/2] && Not[GtQ[c,0]]
```

2: $\int u\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c^2-a^2 d^2=0\ \land\ p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^p\left(c+\frac{d}{x}\right)^p}{\left(1+\frac{c\,x}{d}\right)^p}=0$

Rule: If $c^2-a^2 d^2=0\ \land\ p\notin\mathbb{Z}$, then

$$\int u\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow \frac{x^p\left(c+\frac{d}{x}\right)^p}{\left(1+\frac{c\,x}{d}\right)^p}\int \frac{u}{x^p}\left(1+\frac{c\,x}{d}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_)^p_*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  x^p*(c+d/x)^p/(1+c*x/d)^p*Int[u*(1+c*x/d)^p*E^(n*ArcTanh[a*x])/x^p,x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c^2-a^2*d^2,0] && Not[IntegerQ[p]]
```

4. $\int u\left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2 c+d=0$

1. $\int \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2 c+d=0$

1. $\int \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2 c+d=0\ \land\ p<-1\ \land\ n\notin\mathbb{Z}$

1: $\int \frac{e^{n \operatorname{ArcTanh}[a x]}}{\left(c+d x^2\right)^{3/2}} \, dx$ when $a^2 c + d == 0 \wedge n \notin \mathbb{Z}$

Rule: If $a^2 c + d == 0 \wedge n \notin \mathbb{Z}$, then

$$\int \frac{e^{n \operatorname{ArcTanh}[a x]}}{\left(c+d x^2\right)^{3/2}} \, dx \longrightarrow \frac{(n-a x)\, e^{n \operatorname{ArcTanh}[a x]}}{a c \left(n^2-1\right) \sqrt{c+d x^2}}$$

Program code:

```
Int[E^(n_*ArcTanh[a_.*x_])/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  (n-a*x)*E^(n*ArcTanh[a*x])/(a*c*(n^2-1)*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n]]
```

2: $\int \left(c+d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $a^2 c + d == 0 \wedge p < -1 \wedge n \notin \mathbb{Z} \wedge n^2 - 4 (p+1)^2 \neq 0$

Derivation: ???

Rule: If $a^2 c + d == 0 \wedge p < -1 \wedge n \notin \mathbb{Z} \wedge n^2 - 4 (p+1)^2 \neq 0$, then

$$\int \left(c+d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx \longrightarrow \frac{(n+2 a (p+1) x) \left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTanh}[a x]}}{a c \left(n^2-4 (p+1)^2\right)} - \frac{2 (p+1) (2 p+3)}{c \left(n^2-4 (p+1)^2\right)} \int \left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTanh}[a x]} \, dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  (n+2*a*(p+1)*x)*(c+d*x^2)^(p+1)*E^(n*ArcTanh[a*x])/(a*c*(n^2-4*(p+1)^2)) -
  2*(p+1)*(2*p+3)/(c*(n^2-4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && LtQ[p,-1] && Not[IntegerQ[n]] && NeQ[n^2-4*(p+1)^2,0] && IntegerQ[2*p]
```

2. $\int \left(c+d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $a^2 c + d == 0 \wedge (p \in \mathbb{Z} \vee c > 0)$

1: $\int \frac{e^{n\,\text{ArcTanh}[a\,x]}}{c+d\,x^2}\,dx$ when $a^2\,c+d==0\;\wedge\;\frac{n}{2}\notin\mathbb{Z}$

Rule: If $a^2\,c+d==0\;\wedge\;\frac{n}{2}\notin\mathbb{Z}$, then

$$\int \frac{e^{n\,\text{ArcTanh}[a\,x]}}{c+d\,x^2}\,dx\;\longrightarrow\;\frac{e^{n\,\text{ArcTanh}[a\,x]}}{a\,c\,n}$$

Program code:

```
Int[E^(n_.*ArcTanh[a_.*x_])/(c_+d_.*x_^2),x_Symbol] :=
  E^(n*ArcTanh[a*x])/(a*c*n) /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n/2]]
```

2. $\int (c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c+d==0\;\wedge\;p\in\mathbb{Z}\;\wedge\;\frac{n+1}{2}\in\mathbb{Z}$

1: $\int (c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c+d==0\;\wedge\;p\in\mathbb{Z}\;\wedge\;\frac{n+1}{2}\in\mathbb{Z}^+$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTanh}[z]}==\frac{(1+z)^n}{(1-z^2)^{n/2}}$

Rule: If $a^2\,c+d==0\;\wedge\;p\in\mathbb{Z}\;\wedge\;\frac{n+1}{2}\in\mathbb{Z}^+$, then

$$\int (c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx\;\longrightarrow\;c^p\int (1-a^2\,x^2)^p\,\frac{(1+a\,x)^n}{(1-a^2\,x^2)^{n/2}}\,dx\;\longrightarrow\;c^p\int (1-a^2\,x^2)^{p-\frac{n}{2}}\,(1+a\,x)^n\,dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^p*Int[(1-a^2*x^2)^(p-n/2)*(1+a*x)^n,x] /;
FreeQ[{a,c,d,p},x] && EqQ[a^2*c+d,0] && IntegerQ[p] && IGtQ[(n+1)/2,0] && Not[IntegerQ[p-n/2]]
```

2: $\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} \, dx$ when $a^2 c + d == 0 \land p \in \mathbb{Z} \land \frac{n-1}{2} \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: $e^{n \text{ArcTanh}[z]} == \frac{(1-z^2)^{n/2}}{(1-z)^n}$

Rule: If $a^2 c + d == 0 \land p \in \mathbb{Z} \land \frac{n-1}{2} \in \mathbb{Z}^-$, then

$$\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} \, dx \longrightarrow c^p \int (1 - a^2 x^2)^p \frac{(1 - a^2 x^2)^{n/2}}{(1 - a x)^n} \, dx \longrightarrow c^p \int \frac{(1 - a^2 x^2)^{p+\frac{n}{2}}}{(1 - a x)^n} \, dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_.*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  c^p*Int[(1-a^2*x^2)^(p+n/2)/(1-a*x)^n,x] /;
FreeQ[{a,c,d,p},x] && EqQ[a^2*c+d,0] && IntegerQ[p] && ILtQ[(n-1)/2,0] && Not[IntegerQ[p-n/2]]
```

3: $\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} \, dx$ when $a^2 c + d == 0 \land (p \in \mathbb{Z} \lor c > 0)$

Derivation: Algebraic simplification

Basis: If $a^2 c + d == 0 \land (p \in \mathbb{Z} \lor c > 0)$, then $(c + d x^2)^p == c^p (1 - a x)^p (1 + a x)^p$

Basis: $e^{n \text{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

–

Rule: If $a^2 c + d == 0 \land (p \in \mathbb{Z} \lor c > 0)$, then

$$\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} \, dx \longrightarrow c^p \int (1 - a x)^p (1 + a x)^p \frac{(1 + a x)^{n/2}}{(1 - a x)^{n/2}} \, dx \longrightarrow c^p \int (1 - a x)^{p-\frac{n}{2}} (1 + a x)^{p+\frac{n}{2}} \, dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^p*Int[(1-a*x)^(p-n/2)*(1+a*x)^(p+n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c+d,0] && (IntegerQ[p] || GtQ[c,0])
```

3. $\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} dx$ when $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0)$

1. $\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} dx$ when $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \in \mathbb{Z}$

1: $\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} dx$ when $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \in \mathbb{Z}^+$

Derivation: Algebraic simplification

Basis: $e^{n \text{ArcTanh}[z]} == \frac{(1+z)^n}{(1-z^2)^{n/2}}$

Basis: If $a^2 c + d == 0 \land \frac{n}{2} \in \mathbb{Z}$, then $(1 - a^2 x^2)^{-n/2} == c^{n/2} (c + d x^2)^{-n/2}$

Rule: If $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \in \mathbb{Z}^+$, then

$$\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} dx \longrightarrow \int (c + d x^2)^p \frac{(1 + a x)^n}{(1 - a^2 x^2)^{n/2}} dx \longrightarrow c^{n/2} \int (c + d x^2)^{p-\frac{n}{2}} (1 + a x)^n dx$$

–

Program code:

```
Int[(c_+d_.*x_^2)^p_.*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  c^(n/2)*Int[(c+d*x^2)^(p-n/2)*(1+a*x)^n,x] /;
FreeQ[{a,c,d,p},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[p] || GtQ[c,0]] && IGtQ[n/2,0]
```

2: $\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} dx$ when $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: $e^{n \text{ArcTanh}[z]} == \frac{(1-z^2)^{n/2}}{(1-z)^n}$

Basis: If $a^2 c + d == 0 \land \frac{n}{2} \in \mathbb{Z}$, then $(1 - a^2 x^2)^{n/2} == \frac{1}{c^{n/2}} (c + d x^2)^{n/2}$

Rule: If $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \in \mathbb{Z}^-$, then

$$\int (c + d x^2)^p e^{n \text{ArcTanh}[a x]} dx \to \int (c + d x^2)^p \frac{(1 - a^2 x^2)^{n/2}}{(1 - a x)^n} dx \to \frac{1}{c^{n/2}} \int \frac{(c + d x^2)^{p+\frac{n}{2}}}{(1 - a x)^n} dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_.*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  1/c^(n/2)*Int[(c+d*x^2)^(p+n/2)/(1-a*x)^n,x] /;
FreeQ[{a,c,d,p},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[p] || GtQ[c,0]] && ILtQ[n/2,0]
```

2: $\int \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c+d=0\ \wedge\ \neg\,(p\in\mathbb{Z}\ \vee\ c>0)\ \wedge\ \frac{n}{2}\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $a^2\,c+d=0$, then $\partial_x \frac{\left(c+d\,x^2\right)^p}{\left(1-a^2\,x^2\right)^p}=0$

Rule: If $a^2\,c+d=0\ \wedge\ \neg\,(p\in\mathbb{Z}\ \vee\ c>0)$, then

$$\int \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow \frac{c^{\text{IntPart}[p]}\left(c+d\,x^2\right)^{\text{FracPart}[p]}}{\left(1-a^2\,x^2\right)^{\text{FracPart}[p]}}\int \left(1-a^2\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^IntPart[p]*(c+d*x^2)^FracPart[p]/(1-a^2*x^2)^FracPart[p]*Int[(1-a^2*x^2)^p*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[p] || GtQ[c,0]]
```

2. $\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTanh}[a x]} \, dx$ when $a^2 c + d == 0$

1. $\int x \left(c + d x^2\right)^p e^{n \text{ArcTanh}[a x]} \, dx$ when $a^2 c + d == 0 \land p < -1 \land n \notin \mathbb{Z}$

1: $\int \frac{x \, e^{n \text{ArcTanh}[a x]}}{\left(c + d x^2\right)^{3/2}} \, dx$ when $a^2 c + d == 0 \land n \notin \mathbb{Z}$

Rule: If $a^2 c + d == 0 \land n \notin \mathbb{Z}$, then

$$\int \frac{x \, e^{n \text{ArcTanh}[a x]}}{\left(c + d x^2\right)^{3/2}} \, dx \longrightarrow \frac{(1 - a n x) \, e^{n \text{ArcTanh}[a x]}}{d \left(n^2 - 1\right) \sqrt{c + d x^2}}$$

Program code:

```
Int[x_*E^(n_*ArcTanh[a_.*x_])/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  (1-a*n*x)*E^(n*ArcTanh[a*x])/(d*(n^2-1)*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n]]
```

2: $\int x\left(c+d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $a^2 c+d=0 \wedge p<-1 \wedge n \notin \mathbb{Z}$

- Derivation: Integration by parts

- Basis: $\partial_x \frac{\left(c+d x^2\right)^{p+1}}{2 d (p+1)} = x\left(c+d x^2\right)^p$

- Rule: If $a^2 c+d=0 \wedge p<-1 \wedge n \notin \mathbb{Z}$, then

$$\int x\left(c+d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx \longrightarrow \frac{\left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTanh}[a x]}}{2 d (p+1)}-\frac{a c n}{2 d (p+1)} \int\left(c+d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$$

$$\longrightarrow -\frac{(2 (p+1)+a n x)\left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTanh}[a x]}}{d\left(n^2-4 (p+1)^2\right)}-\frac{n (2 p+3)}{a c\left(n^2-4 (p+1)^2\right)} \int\left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTanh}[a x]} \, dx$$

- Program code:

```
Int[x_*(c_+d_.*x_^2)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  (c+d*x^2)^(p+1)*E^(n*ArcTanh[a*x])/(2*d*(p+1)) - a*c*n/(2*d*(p+1))*Int[(c+d*x^2)^p*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && LtQ[p,-1] && Not[IntegerQ[n]] && IntegerQ[2*p]
```

```
(* Int[x_*(c_+d_.*x_^2)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  -(2*(p+1)+a*n*x)*(c+d*x^2)^(p+1)*E^(n*ArcTanh[a*x])/(d*(n^2-4*(p+1)^2)) -
  n*(2*p+3)/(a*c*(n^2-4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && LeQ[p,-1] && NeQ[n^2-4*(p+1)^2,0] && Not[IntegerQ[n]] *)
```

2. $\int x^2\left(c+d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $a^2 c+d=0 \wedge p<-1 \wedge n \notin \mathbb{Z}$

1: $\int x^2\left(c+d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $a^2 c+d=0 \wedge n^2+2 (p+1)=0 \wedge n \notin \mathbb{Z}$

Rule: If $a^2 c+d=0 \wedge n^2+2 (p+1)=0 \wedge n \notin \mathbb{Z}$, then

$$\int x^2 \left(c + d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx \longrightarrow \frac{(1 - a n x) \left(c + d x^2\right)^{p+1} e^{n \operatorname{ArcTanh}[a x]}}{a d n \left(n^2 - 1\right)}$$

Program code:

```
Int[x_^2*(c_+d_.*x_^2)^p_.*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  (1-a*n*x)*(c+d*x^2)^(p+1)*E^(n*ArcTanh[a*x])/(a*d*n*(n^2-1)) /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && EqQ[n^2+2*(p+1),0] && Not[IntegerQ[n]]
```

2: $\int x^2 \left(c + d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $a^2 c + d == 0 \land p < -1 \land n \notin \mathbb{Z} \land n^2 - 4 (p + 1)^2 \neq 0$

Derivation: Algebraic expansion and ???

Basis: $x^2 \left(c + d x^2\right)^p == -\frac{c \left(c+d x^2\right)^p}{d} + \frac{\left(c+d x^2\right)^{p+1}}{d}$

Rule: If $a^2 c + d == 0 \land p < -1 \land n \notin \mathbb{Z} \land n^2 - 4 (p + 1)^2 \neq 0$, then

$$\int x^2 \left(c + d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx \longrightarrow -\frac{c}{d} \int \left(c + d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx + \frac{1}{d} \int \left(c + d x^2\right)^{p+1} e^{n \operatorname{ArcTanh}[a x]} \, dx$$

$$\longrightarrow -\frac{(n + 2 (p + 1) a x) \left(c + d x^2\right)^{p+1} e^{n \operatorname{ArcTanh}[a x]}}{a d \left(n^2 - 4 (p + 1)^2\right)} + \frac{n^2 + 2 (p + 1)}{d \left(n^2 - 4 (p + 1)^2\right)} \int \left(c + d x^2\right)^{p+1} e^{n \operatorname{ArcTanh}[a x]} \, dx$$

Program code:

```
Int[x_^2*(c_+d_.*x_^2)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  -(n+2*(p+1)*a*x)*(c+d*x^2)^(p+1)*E^(n*ArcTanh[a*x])/(a*d*(n^2-4*(p+1)^2)) +
  (n^2+2*(p+1))/(d*(n^2-4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && LtQ[p,-1] && Not[IntegerQ[n]] && NeQ[n^2-4*(p+1)^2,0] && IntegerQ[2*p]
```

3. $\int x^m \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2 c+d=0 \land (p\in\mathbb{Z} \lor c>0)$

1. $\int x^m \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2 c+d=0 \land (p\in\mathbb{Z} \lor c>0) \land \frac{n+1}{2}\in\mathbb{Z}$

1: $\int x^m \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2 c+d=0 \land (p\in\mathbb{Z} \lor c>0) \land \frac{n+1}{2}\in\mathbb{Z}^+$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTanh}[z]} = \frac{(1+z)^n}{\left(1-z^2\right)^{n/2}}$

Rule: If $a^2 c+d=0 \land (p\in\mathbb{Z} \lor c>0) \land \frac{n+1}{2}\in\mathbb{Z}^+$, then

$$\int x^m \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow c^p \int x^m \left(1-a^2 x^2\right)^p \frac{(1+a\,x)^n}{\left(1-a^2 x^2\right)^{n/2}}\,dx \longrightarrow c^p \int x^m \left(1-a^2 x^2\right)^{p-\frac{n}{2}} (1+a\,x)^n\,dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^p*Int[x^m*(1-a^2*x^2)^(p-n/2)*(1+a*x)^n,x] /;
FreeQ[{a,c,d,m,p},x] && EqQ[a^2*c+d,0] && (IntegerQ[p] || GtQ[c,0]) && IGtQ[(n+1)/2,0] && Not[IntegerQ[p-n/2]]
```

2: $\int x^m \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c+d==0\ \land\ (p\in\mathbb{Z}\ \lor\ c>0)\ \land\ \frac{n-1}{2}\in\mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTanh}[z]}==\frac{\left(1-z^2\right)^{n/2}}{(1-z)^n}$

Rule: If $a^2\,c+d==0\ \land\ (p\in\mathbb{Z}\ \lor\ c>0)\ \land\ \frac{n-1}{2}\in\mathbb{Z}^-$, then

$$\int x^m \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx\ \longrightarrow\ c^p\int x^m\left(1-a^2\,x^2\right)^p\frac{\left(1-a^2\,x^2\right)^{n/2}}{(1-a\,x)^n}\,dx\ \longrightarrow\ c^p\int\frac{x^m\left(1-a^2\,x^2\right)^{p+\frac{n}{2}}}{(1-a\,x)^n}\,dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_.*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  c^p*Int[x^m*(1-a^2*x^2)^(p+n/2)/(1-a*x)^n,x] /;
FreeQ[{a,c,d,m,p},x] && EqQ[a^2*c+d,0] && (IntegerQ[p] || GtQ[c,0]) && ILtQ[(n-1)/2,0] && Not[IntegerQ[p-n/2]]
```

2: $\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $a^2 c + d == 0 \land (p \in \mathbb{Z} \lor c > 0)$

Derivation: Algebraic simplification

Basis: If $a^2 c + d == 0 \land (p \in \mathbb{Z} \lor c > 0)$, then $\left(c + d x^2\right)^p == c^p (1 - a x)^p (1 + a x)^p$

Basis: $e^{n \operatorname{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

–

Rule: If $a^2 c + d == 0 \land (p \in \mathbb{Z} \lor c > 0)$, then

$$\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx \longrightarrow c^p \int x^m (1 - a x)^p (1 + a x)^p \frac{(1 + a x)^{n/2}}{(1 - a x)^{n/2}} \, dx \longrightarrow c^p \int x^m (1 - a x)^{p - \frac{n}{2}} (1 + a x)^{p + \frac{n}{2}} \, dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^p*Int[x^m*(1-a*x)^(p-n/2)*(1+a*x)^(p+n/2),x] /;
FreeQ[{a,c,d,m,n,p},x] && EqQ[a^2*c+d,0] && (IntegerQ[p] || GtQ[c,0])
```

4. $\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTanh}[a x]} dx$ when $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0)$

   1. $\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTanh}[a x]} dx$ when $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \in \mathbb{Z}$

      1: $\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTanh}[a x]} dx$ when $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \in \mathbb{Z}^+$

Derivation: Algebraic simplification

Basis: $e^{n \text{ArcTanh}[z]} == \frac{(1+z)^n}{\left(1-z^2\right)^{n/2}}$

Basis: If $a^2 c + d == 0 \land \frac{n}{2} \in \mathbb{Z}$, then $\left(1 - a^2 x^2\right)^{-n/2} == c^{n/2} \left(c + d x^2\right)^{-n/2}$

Rule: If $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \in \mathbb{Z}^+$, then

$$\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTanh}[a x]} dx \to \int x^m \left(c + d x^2\right)^p \frac{(1 + a x)^n}{\left(1 - a^2 x^2\right)^{n/2}} dx \to c^{n/2} \int x^m \left(c + d x^2\right)^{p-\frac{n}{2}} (1 + a x)^n dx$$

–

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_.*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  c^(n/2)*Int[x^m*(c+d*x^2)^(p-n/2)*(1+a*x)^n,x] /;
FreeQ[{a,c,d,m,p},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[p] || GtQ[c,0]] && IGtQ[n/2,0]
```

2: $\int x^m \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2 c+d==0\ \land\ \neg\ (p\in\mathbb{Z}\ \lor\ c>0)\ \land\ \frac{n}{2}\in\mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTanh}[z]} == \frac{\left(1-z^2\right)^{n/2}}{(1-z)^n}$

Basis: If $a^2 c+d==0\ \land\ \frac{n}{2}\in\mathbb{Z}$, then $\left(1-a^2 x^2\right)^{n/2} == \frac{1}{c^{n/2}}\left(c+d\,x^2\right)^{n/2}$

Rule: If $a^2 c+d==0\ \land\ \neg\ (p\in\mathbb{Z}\ \lor\ c>0)\ \land\ \frac{n}{2}\in\mathbb{Z}^-$, then

$$\int x^m \left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow \int x^m \left(c+d\,x^2\right)^p \frac{\left(1-a^2 x^2\right)^{n/2}}{(1-a\,x)^n}\,dx \longrightarrow \frac{1}{c^{n/2}}\int \frac{x^m \left(c+d\,x^2\right)^{p+\frac{n}{2}}}{(1-a\,x)^n}\,dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_.*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  1/c^(n/2)*Int[x^m*(c+d*x^2)^(p+n/2)/(1-a*x)^n,x] /;
FreeQ[{a,c,d,m,p},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[p] || GtQ[c,0]] && ILtQ[n/2,0]
```

2: $\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $a^2 c + d == 0$, then $\partial_x \frac{\left(c+d x^2\right)^p}{\left(1-a^2 x^2\right)^p} == 0$

Rule: If $a^2 c + d == 0 \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{n}{2} \notin \mathbb{Z}$, then

$$\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx \longrightarrow \frac{c^{\operatorname{IntPart}[p]} \left(c + d x^2\right)^{\operatorname{FracPart}[p]}}{\left(1 - a^2 x^2\right)^{\operatorname{FracPart}[p]}} \int x^m \left(1 - a^2 x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^IntPart[p]*(c+d*x^2)^FracPart[p]/(1-a^2*x^2)^FracPart[p]*Int[x^m*(1-a^2*x^2)^p*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,m,n,p},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[p] || GtQ[c,0]] && Not[IntegerQ[n/2]]
```

3. $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c+d==0$

1: $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c+d==0\,\land\,(p\in\mathbb{Z}\,\lor\,c>0)$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTanh}[z]}==\frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Basis: $(1-z^2)^p==(1-z)^p\,(1+z)^p$

Rule: If $a^2\,c+d==0\,\land\,(p\in\mathbb{Z}\,\lor\,c>0)$, then

$$\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx\,\longrightarrow\,c^p\int u\,(1-a\,x)^p\,(1+a\,x)^p\,\frac{(1+a\,x)^{n/2}}{(1-a\,x)^{n/2}}\,dx\,\longrightarrow\,c^p\int u\,(1-a\,x)^{p-\frac{n}{2}}\,(1+a\,x)^{p+\frac{n}{2}}\,dx$$

Program code:

```
Int[u_*(c_+d_.*x_^2)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^p*Int[u*(1-a*x)^(p-n/2)*(1+a*x)^(p+n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c+d,0] && (IntegerQ[p] || GtQ[c,0])
```

2. $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c+d==0\,\land\,\neg\,(p\in\mathbb{Z}\,\lor\,c>0)$

1: $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c+d==0\,\land\,\neg\,(p\in\mathbb{Z}\,\lor\,c>0)\,\land\,\frac{n}{2}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $a^2\,c+d==0$, then $\partial_x\,\frac{(c+d\,x^2)^p}{(1-a\,x)^p\,(1+a\,x)^p}==0$

Basis: $e^{n\,\text{ArcTanh}[z]}==\frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Rule: If $a^2\,c+d==0\,\land\,\neg\,(p\in\mathbb{Z}\,\lor\,c>0)\,\land\,\frac{n}{2}\in\mathbb{Z}$, then

$$\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx\,\longrightarrow\,\frac{c^{\text{IntPart}[p]}\,(c+d\,x^2)^{\text{FracPart}[p]}}{(1-a\,x)^{\text{FracPart}[p]}\,(1+a\,x)^{\text{FracPart}[p]}}\int u\,(1-a\,x)^{p-\frac{n}{2}}\,(1+a\,x)^{p+\frac{n}{2}}\,dx$$

Program code:

```
Int[u_*(c_+d_.*x_^2)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  c^IntPart[p]*(c+d*x^2)^FracPart[p]/((1-a*x)^FracPart[p]*(1+a*x)^FracPart[p])*
    Int[u*(1-a*x)^(p-n/2)*(1+a*x)^(p+n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[p] || GtQ[c,0]] && IntegerQ[n/2]
```

2: $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $a^2\,c+d==0\,\land\,\neg\,(p\in\mathbb{Z}\,\lor\,c>0)\,\land\,\frac{n}{2}\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $a^2\,c+d==0$, then $\partial_x\,\frac{(c+d\,x^2)^p}{(1-a^2\,x^2)^p}==0$

Rule: If $a^2\,c+d==0\,\land\,\neg\,(p\in\mathbb{Z}\,\lor\,c>0)\,\land\,\frac{n}{2}\notin\mathbb{Z}$, then

$$\int u\left(c+d\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx\;\longrightarrow\;\frac{c^{\text{IntPart}[p]}\left(c+d\,x^2\right)^{\text{FracPart}[p]}}{\left(1-a^2\,x^2\right)^{\text{FracPart}[p]}}\int u\left(1-a^2\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$$

Program code:

```
Int[u_*(c_+d_.*x_^2)^p_*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  c^IntPart[p]*(c+d*x^2)^FracPart[p]/(1-a^2*x^2)^FracPart[p]*Int[u*(1-a^2*x^2)^p*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[p] || GtQ[c,0]] && Not[IntegerQ[n/2]]
```

5. $\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c+a^2\,d==0$

1: $\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c+a^2\,d==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c+a^2\,d==0\,\land\,p\in\mathbb{Z}$, then $\left(c+\frac{d}{x^2}\right)^p==\frac{d^p}{x^{2\,p}}\left(1-a^2\,x^2\right)^p$

Rule: If $c+a^2\,d==0\,\land\,p\in\mathbb{Z}$, then

$$\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx\;\longrightarrow\;d^p\int\frac{u}{x^{2\,p}}\left(1-a^2\,x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_.*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  d^p*Int[u/x^(2*p)*(1-a^2*x^2)^p*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[c+a^2*d,0] && IntegerQ[p]
```

2. $\int u \left(c + \frac{d}{x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c + a^2 d == 0 \land p \notin \mathbb{Z}$

1. $\int u \left(c + \frac{d}{x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c + a^2 d == 0 \land p \notin \mathbb{Z} \land \frac{n}{2} \in \mathbb{Z}$

1: $\int u \left(c + \frac{d}{x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c + a^2 d == 0 \land p \notin \mathbb{Z} \land \frac{n}{2} \in \mathbb{Z} \land c > 0$

Derivation: Algebraic simplification

Basis: $\left(1 - z^2\right)^p == (1 - z)^p\,(1 + z)^p$

Rule: If $c + a^2 d == 0 \land p \notin \mathbb{Z} \land \frac{n}{2} \in \mathbb{Z} \land c > 0$, then

$$\int u \left(c + \frac{d}{x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow c^p \int u \left(1 - \frac{1}{a^2 x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow c^p \int u \left(1 - \frac{1}{a\,x}\right)^p \left(1 + \frac{1}{a\,x}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  c^p*Int[u*(1-1/(a*x))^p*(1+1/(a*x))^p*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,p},x] && EqQ[c+a^2*d,0] && Not[IntegerQ[p]] && IntegerQ[n/2] && GtQ[c,0]
```

2: $\int u \left(c+\frac{d}{x^2}\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $c+a^2 d=0 \land p\notin\mathbb{Z} \land \frac{n}{2}\in\mathbb{Z} \land c \ngtr 0$

Derivation: Piecewise constant extraction

Basis: If $c+a^2 d=0$, then $\partial_x \frac{x^{2p}\left(c+\frac{d}{x^2}\right)^p}{(1-a x)^p (1+a x)^p}=0$

Rule: If $c+a^2 d=0 \land p\notin\mathbb{Z} \land \frac{n}{2}\in\mathbb{Z} \land c \ngtr 0$, then

$$\int u \left(c+\frac{d}{x^2}\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx \longrightarrow \frac{x^{2p}\left(c+\frac{d}{x^2}\right)^p}{(1-a x)^p (1+a x)^p} \int \frac{u}{x^{2p}} (1-a x)^p (1+a x)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  x^(2*p)*(c+d/x^2)^p/((1-a*x)^p*(1+a*x)^p)*Int[u/x^(2*p)*(1-a*x)^p*(1+a*x)^p*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c+a^2*d,0] && Not[IntegerQ[p]] && IntegerQ[n/2] && Not[GtQ[c,0]]
```

2: $\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$ when $c+a^2 d==0 \land p\notin\mathbb{Z} \land \frac{n}{2}\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c+a^2 d==0$, then $\partial_x \frac{x^{2p}\left(c+\frac{d}{x^2}\right)^p}{\left(1-a^2 x^2\right)^p}==0$

Rule: If $c+a^2 d==0 \land p\notin\mathbb{Z} \land \frac{n}{2}\notin\mathbb{Z}$, then

$$\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow \frac{x^{2p}\left(c+\frac{d}{x^2}\right)^p}{\left(1-a^2 x^2\right)^p}\int \frac{u}{x^{2p}}\left(1-a^2 x^2\right)^p e^{n\,\text{ArcTanh}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_*E^(n_.*ArcTanh[a_.*x_]),x_Symbol] :=
  x^(2*p)*(c+d/x^2)^p/(1+c*x^2/d)^p*Int[u/x^(2*p)*(1+c*x^2/d)^p*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c+a^2*d,0] && Not[IntegerQ[p]] && Not[IntegerQ[n/2]]
```

2. $\int u\, e^{n\,\text{ArcTanh}[a+b\,x]}\,dx$

1: $\int e^{n\,\text{ArcTanh}[c\,(a+b\,x)]}\,dx$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

–

Rule:

$$\int e^{n\,\text{ArcTanh}[c\,(a+b\,x)]}\,dx \longrightarrow \int \frac{(1+a\,c+b\,c\,x)^{n/2}}{(1-a\,c-b\,c\,x)^{n/2}}\,dx$$

–

Program code:

```
Int[E^(n_.*ArcTanh[c_.*(a_+b_.*x_)]),x_Symbol] :=
  Int[(1+a*c+b*c*x)^(n/2)/(1-a*c-b*c*x)^(n/2),x] /;
FreeQ[{a,b,c,n},x]
```

2. $\int (d+e\,x)^m\,e^{n\,\text{ArcTanh}[c\,(a+b\,x)]}\,dx$

1: $\int x^m\,e^{n\,\text{ArcTanh}[c\,(a+b\,x)]}\,dx$ when $m\in\mathbb{Z}^-\land -1<n<1$

Derivation: Algebraic simplification and integration by substitution

Basis: $e^{n\,\text{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Basis: If $m\in\mathbb{Z}\land -1<n<1$, then

$$x^m\,\frac{(1+c\,(a+b\,x))^{n/2}}{(1-c\,(a+b\,x))^{n/2}} == \frac{4}{n\,b^{m+1}\,c^{m+1}}\,\text{Subst}\left[\frac{x^{2/n}\left(-1-a\,c+(1-a\,c)\,x^{2/n}\right)^m}{\left(1+x^{2/n}\right)^{m+2}},\,x,\,\frac{(1+c\,(a+b\,x))^{n/2}}{(1-c\,(a+b\,x))^{n/2}}\right]\partial_x\frac{(1+c\,(a+b\,x))^{n/2}}{(1-c\,(a+b\,x))^{n/2}}$$

Note: There should be an algebraic substitution rule that makes this rule redundant.

Rule: If $m\in\mathbb{Z}^-\land -1<n<1$, then

$$\int x^m\,e^{n\,\text{ArcTanh}[c\,(a+b\,x)]}\,dx \to \int x^m\,\frac{(1+c\,(a+b\,x))^{n/2}}{(1-c\,(a+b\,x))^{n/2}}\,dx$$
$$\to \frac{4}{n\,b^{m+1}\,c^{m+1}}\,\text{Subst}\left[\int\frac{x^{2/n}\left(-1-a\,c+(1-a\,c)\,x^{2/n}\right)^m}{\left(1+x^{2/n}\right)^{m+2}}\,dx,\,x,\,\frac{(1+c\,(a+b\,x))^{n/2}}{(1-c\,(a+b\,x))^{n/2}}\right]$$

Program code:

```
Int[x_^m_*E^(n_*ArcTanh[c_.*(a_+b_.*x_)]),x_Symbol] :=
  4/(n*b^(m+1)*c^(m+1))*
    Subst[Int[x^(2/n)*(-1-a*c+(1-a*c)*x^(2/n))^m/(1+x^(2/n))^(m+2),x],x,(1+c*(a+b*x))^(n/2)/(1-c*(a+b*x))^(n/2)] /;
FreeQ[{a,b,c},x] && ILtQ[m,0] && LtQ[-1,n,1]
```

2: $\int (d+e x)^m e^{n \text{ArcTanh}[c (a+b x)]} \, dx$

Derivation: Algebraic simplification

Basis: $e^{n \text{ArcTanh}[z]} = \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Rule:

$$\int (d+e x)^m e^{n \text{ArcTanh}[c (a+b x)]} \, dx \longrightarrow \int (d+e x)^m \frac{(1+a c+b c x)^{n/2}}{(1-a c-b c x)^{n/2}} \, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*E^(n_.*ArcTanh[c_.*(a_+b_.*x_)]),x_Symbol] :=
  Int[(d+e*x)^m*(1+a*c+b*c*x)^(n/2)/(1-a*c-b*c*x)^(n/2),x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

3. $\int u \left(c+d x+e x^2\right)^p e^{n \text{ArcTanh}[a+b x]} \, dx$ when $b d = 2 a e \land b^2 c + e \left(1-a^2\right) = 0$

1: $\int u \left(c+d x+e x^2\right)^p e^{n \text{ArcTanh}[a+b x]} \, dx$ when $b d = 2 a e \land b^2 c + e \left(1-a^2\right) = 0 \land \left(p \in \mathbb{Z} \lor \frac{c}{1-a^2} > 0\right)$

Derivation: Algebraic simplification

Basis: If $b d = 2 a e \land b^2 c + e \left(1-a^2\right) = 0$, then $c + d x + e x^2 = \frac{c}{1-a^2} \left(1-(a+b x)^2\right)$

Basis: $\left(1-z^2\right)^p = (1-z)^p (1+z)^p$

Basis: $e^{n \text{ArcTanh}[z]} = \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Rule: If $b d = 2 a e \land b^2 c + e \left(1-a^2\right) = 0 \land \left(p \in \mathbb{Z} \lor \frac{c}{1-a^2} > 0\right)$, then

$$\int u \left(c+d x+e x^2\right)^p e^{n \text{ArcTanh}[a+b x]} \, dx \longrightarrow \left(\frac{c}{1-a^2}\right)^p \int u \left(1-(a+b x)^2\right)^p e^{n \text{ArcTanh}[a+b x]} \, dx$$

$$\rightarrow \left(\frac{c}{1-a^2}\right)^p \int u\,(1-a-b\,x)^p\,(1+a+b\,x)^p\,\frac{(1+a+b\,x)^{n/2}}{(1-a-b\,x)^{n/2}}\,dx$$

$$\rightarrow \left(\frac{c}{1-a^2}\right)^p \int u\,(1-a-b\,x)^{p-n/2}\,(1+a+b\,x)^{p+n/2}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_+e_.*x_^2)^p_.*E^(n_.*ArcTanh[a_+b_.*x_]),x_Symbol] :=
  (c/(1-a^2))^p*Int[u*(1-a-b*x)^(p-n/2)*(1+a+b*x)^(p+n/2),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[b*d-2*a*e,0] && EqQ[b^2*c+e(1-a^2),0] && (IntegerQ[p] || GtQ[c/(1-a^2),0])
```

2: $\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcTanh}[a+b\,x]}\,dx$ when $b\,d==2\,a\,e\wedge b^2\,c+e\left(1-a^2\right)==0\wedge\neg\left(p\in\mathbb{Z}\vee\frac{c}{1-a^2}>0\right)$

Derivation: Piecewise constant extraction

Basis: If $b\,d==2\,a\,e\wedge b^2\,c+e\left(1-a^2\right)==0$, then $\partial_x\frac{\left(c+d\,x+e\,x^2\right)^p}{\left(1-a^2-2\,a\,b\,x-b^2\,x^2\right)^p}==0$

Rule: If $b\,d==2\,a\,e\wedge b^2\,c+e\left(1-a^2\right)==0\wedge\neg\left(p\in\mathbb{Z}\vee\frac{c}{1-a^2}>0\right)$, then

$$\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcTanh}[a+b\,x]}\,dx\rightarrow\frac{\left(c+d\,x+e\,x^2\right)^p}{\left(1-a^2-2\,a\,b\,x-b^2\,x^2\right)^p}\int u\left(1-a^2-2\,a\,b\,x-b^2\,x^2\right)^p e^{n\,\text{ArcTanh}[a+b\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_+e_.*x_^2)^p_.*E^(n_.*ArcTanh[a_+b_.*x_]),x_Symbol] :=
  (c+d*x+e*x^2)^p/(1-a^2-2*a*b*x-b^2*x^2)^p*Int[u*(1-a^2-2*a*b*x-b^2*x^2)^p*E^(n*ArcTanh[a*x]),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[b*d-2*a*e,0] && EqQ[b^2*c+e(1-a^2),0] && Not[IntegerQ[p] || GtQ[c/(1-a^2),0]]
```

3: $\int u\, e^{n\,\text{ArcTanh}\left[\frac{c}{a+b\,x}\right]}\, dx$

Derivation: Algebraic simplification

Basis: $\text{ArcTanh}[z] == \text{ArcCoth}\left[\frac{1}{z}\right]$

Rule:

$$\int u\, e^{n\,\text{ArcTanh}\left[\frac{c}{a+b\,x}\right]}\, dx \longrightarrow \int u\, e^{n\,\text{ArcCoth}\left[\frac{a}{c}+\frac{b\,x}{c}\right]}\, dx$$

Program code:

```
Int[u_.*E^(n_.*ArcTanh[c_./(a_.+b_.*x_)]),x_Symbol] :=
  Int[u*E^(n*ArcCoth[a/c+b*x/c]),x] /;
FreeQ[{a,b,c,n},x]
```

Rules for integrands involving exponentials of inverse hyperbolic cotangents

1. $\int u\, e^{n \operatorname{ArcCoth}[a x]}\, dx$

1: $\int u\, e^{n \operatorname{ArcCoth}[a x]}\, dx$ when $\frac{n}{2} \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $\frac{n}{2} \in \mathbb{Z}$, then $e^{n \operatorname{ArcCoth}[z]} == (-1)^{n/2}\, e^{n \operatorname{ArcTanh}[z]}$

Rule: If $\frac{n}{2} \in \mathbb{Z}$, then

$$\int u\, e^{n \operatorname{ArcCoth}[a x]}\, dx \longrightarrow (-1)^{n/2} \int u\, e^{n \operatorname{ArcTanh}[a x]}\, dx$$

Program code:

```
Int[u_.*E^(n_*ArcCoth[a_.*x_]),x_Symbol] :=
  (-1)^(n/2)*Int[u*E^(n*ArcTanh[a*x]),x] /;
FreeQ[a,x] && IntegerQ[n/2]
```

2. $\int u\, e^{n \operatorname{ArcCoth}[a x]}\, dx$ when $\frac{n}{2} \notin \mathbb{Z}$

1. $\int x^m\, e^{n \operatorname{ArcCoth}[a x]}\, dx$ when $\frac{n}{2} \notin \mathbb{Z}$

1. $\int x^m\, e^{n \operatorname{ArcCoth}[a x]}\, dx$ when $\frac{n}{2} \notin \mathbb{Z} \land m \in \mathbb{Z}$

1: $\int x^m\, e^{n \operatorname{ArcCoth}[a x]}\, dx$ when $\frac{n-1}{2} \in \mathbb{Z} \land m \in \mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{\frac{n+1}{2}}}{\left(1-\frac{1}{z}\right)^{\frac{n-1}{2}}\sqrt{1-\frac{1}{z^2}}}$

Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule: If $\frac{n-1}{2} \in \mathbb{Z} \land m \in \mathbb{Z}$, then

$$\int x^m e^{n\,\text{ArcCoth}[a x]}\,dx \longrightarrow \int \frac{\left(1+\frac{1}{a x}\right)^{\frac{n+1}{2}}}{\left(\frac{1}{x}\right)^m \left(1-\frac{1}{a x}\right)^{\frac{n-1}{2}}\sqrt{1-\frac{1}{a^2 x^2}}}\,dx \longrightarrow -\text{Subst}\left[\int \frac{\left(1+\frac{x}{a}\right)^{\frac{n+1}{2}}}{x^{m+2}\left(1-\frac{x}{a}\right)^{\frac{n-1}{2}}\sqrt{1-\frac{x^2}{a^2}}}\,dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -Subst[Int[(1+x/a)^((n+1)/2)/(x^2*(1-x/a)^((n-1)/2)*Sqrt[1-x^2/a^2]),x],x,1/x] /;
FreeQ[a,x] && IntegerQ[(n-1)/2]
```

```
Int[x_^m_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -Subst[Int[(1+x/a)^((n+1)/2)/(x^(m+2)*(1-x/a)^((n-1)/2)*Sqrt[1-x^2/a^2]),x],x,1/x] /;
FreeQ[a,x] && IntegerQ[(n-1)/2] && IntegerQ[m]
```

2: $\int x^m e^{n\,\text{ArcCoth}[a x]}\,dx$ when $n \notin \mathbb{Z} \land m \in \mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule: If $n \notin \mathbb{Z} \land m \in \mathbb{Z}$, then

$$\int x^m e^{n\,\text{ArcCoth}[a\,x]}\,dx \longrightarrow \int \frac{\left(1+\frac{1}{a\,x}\right)^{n/2}}{\left(\frac{1}{x}\right)^m \left(1-\frac{1}{a\,x}\right)^{n/2}}\,dx \longrightarrow -\text{Subst}\left[\int \frac{\left(1+\frac{x}{a}\right)^{n/2}}{x^{m+2}\left(1-\frac{x}{a}\right)^{n/2}}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[E^(n_*ArcCoth[a_.*x_]),x_Symbol] :=
  -Subst[Int[(1+x/a)^(n/2)/(x^2*(1-x/a)^(n/2)),x],x,1/x] /;
FreeQ[{a,n},x] && Not[IntegerQ[n]]
```

```
Int[x_^m_.*E^(n_*ArcCoth[a_.*x_]),x_Symbol] :=
  -Subst[Int[(1+x/a)^(n/2)/(x^(m+2)*(1-x/a)^(n/2)),x],x,1/x] /;
FreeQ[{a,n},x] && Not[IntegerQ[n]] && IntegerQ[m]
```

2. $\int x^m e^{n\,\text{ArcCoth}[a x]}\,dx$ when $\frac{n}{2}\notin\mathbb{Z}\land m\notin\mathbb{Z}$

1: $\int x^m e^{n\,\text{ArcCoth}[a x]}\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}\land m\notin\mathbb{Z}$

Derivation: Algebraic simplification, piecewise constant extraction and integration by substitution!

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{\frac{n+1}{2}}}{\left(1-\frac{1}{z}\right)^{\frac{n-1}{2}}\sqrt{1-\frac{1}{z^2}}}$

Basis: $\partial_x\left(x^m\left(\frac{1}{x}\right)^m\right) == 0$

Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule: If $\frac{n-1}{2}\in\mathbb{Z}\land m\notin\mathbb{Z}$, then

$$\int x^m e^{n\,\text{ArcCoth}[a x]}\,dx \to x^m\left(\frac{1}{x}\right)^m\int\frac{\left(1+\frac{1}{a x}\right)^{\frac{n+1}{2}}}{\left(\frac{1}{x}\right)^m\left(1-\frac{1}{a x}\right)^{\frac{n-1}{2}}\sqrt{1-\frac{1}{a^2 x^2}}}\,dx \to -x^m\left(\frac{1}{x}\right)^m\text{Subst}\left[\int\frac{\left(1+\frac{x}{a}\right)^{\frac{n+1}{2}}}{x^{m+2}\left(1-\frac{x}{a}\right)^{\frac{n-1}{2}}\sqrt{1-\frac{x^2}{a^2}}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[x_^m_*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -x^m*(1/x)^m*Subst[Int[(1+x/a)^((n+1)/2)/(x^(m+2)*(1-x/a)^((n-1)/2)*Sqrt[1-x^2/a^2]),x],x,1/x] /;
FreeQ[{a,m},x] && IntegerQ[(n-1)/2] && Not[IntegerQ[m]]
```

2: $\int x^m e^{n\,\text{ArcCoth}[a x]}\,dx$ when $n\notin\mathbb{Z}\land m\notin\mathbb{Z}$

Derivation: Algebraic simplification, piecewise constant extraction and integration by substitution!

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $\partial_x \left(x^m \left(\frac{1}{x}\right)^m\right) == 0$

Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

- Rule: If $n \notin \mathbb{Z} \land m \notin \mathbb{Z}$, then

$$\int x^m e^{n\,\text{ArcCoth}[a\,x]}\,dx \longrightarrow x^m \left(\frac{1}{x}\right)^m \int \frac{\left(1+\frac{1}{a\,x}\right)^{n/2}}{\left(\frac{1}{x}\right)^m \left(1-\frac{1}{a\,x}\right)^{n/2}}\,dx \longrightarrow -x^m \left(\frac{1}{x}\right)^m \text{Subst}\left[\int \frac{\left(1+\frac{x}{a}\right)^{n/2}}{x^{m+2}\left(1-\frac{x}{a}\right)^{n/2}}\,dx, x, \frac{1}{x}\right]$$

- Program code:

```
Int[x_^m_*E^(n_*ArcCoth[a_.*x_]),x_Symbol] :=
  -x^m*(1/x)^m*Subst[Int[(1+x/a)^(n/2)/(x^(m+2)*(1-x/a)^(n/2)),x],x,1/x] /;
FreeQ[{a,m,n},x] && Not[IntegerQ[n]] && Not[IntegerQ[m]]
```

2. $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $a^2 c^2 - d^2 == 0 \land \frac{n}{2} \notin \mathbb{Z}$

1: $\int (c+d\,x)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $a\,c + d == 0 \land p == \frac{n}{2} \notin \mathbb{Z}$

- Rule: If $a\,c + d == 0 \land p == \frac{n}{2} \notin \mathbb{Z}$, then

$$\int (c+d\,x)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx \longrightarrow \frac{(1+a\,x)\,(c+d\,x)^p\,e^{n\,\text{ArcCoth}[a\,x]}}{a\,(p+1)}$$

- Program code:

```
Int[(c_+d_.*x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  (1+a*x)*(c+d*x)^p*E^(n*ArcCoth[a*x])/(a*(p+1)) /;
FreeQ[{a,c,d,n,p},x] && EqQ[a*c+d,0] && EqQ[p,n/2] && Not[IntegerQ[n/2]]
```

x. $\int x^m (c+d x)^p e^{n \text{ArcCoth}[a x]} dx$ when $a c+d==0 \wedge \frac{n-1}{2} \in \mathbb{Z} \wedge m \in \mathbb{Z}$

1: $\int x^m (c+d x)^p e^{n \text{ArcCoth}[a x]} dx$ when $a c+d==0 \wedge \frac{n-1}{2} \in \mathbb{Z} \wedge m \in \mathbb{Z} \wedge p \in \mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: If $n \in \mathbb{Z}$, then $e^{n \text{ArcCoth}[a x]} == (-a)^n c^n x^n (c-a c x)^{-n} \left(1-\frac{1}{a^2 x^2}\right)^{n/2}$

Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule: If $a c+d==0 \wedge \frac{n-1}{2} \in \mathbb{Z} \wedge m \in \mathbb{Z} \wedge p \in \mathbb{Z}$, then

$$\int x^m (c+d x)^p e^{n \text{ArcCoth}[a x]} dx \longrightarrow (-a)^n c^n \int x^{m+n} (c+d x)^{p-n} \left(1-\frac{1}{a^2 x^2}\right)^{n/2} dx \longrightarrow -(-a)^n c^n \text{Subst}\left[\int \frac{(d+c x)^{p-n} \left(1-\frac{x^2}{a^2}\right)^{n/2}}{x^{m+p+2}} dx, x, \frac{1}{x}\right]$$

Program code:

```
(* Int[(c_+d_.*x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -(-a)^n*c^n*Subst[Int[(d+c*x)^(p-n)*(1-x^2/a^2)^(n/2)/x^(p+2),x],x,1/x] /;
FreeQ[{a,c,d},x] && EqQ[a*c+d,0] && IntegerQ[(n-1)/2] && IntegerQ[p] *)
```

```
(* Int[x_^m_.*(c_+d_.*x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -(-a)^n*c^n*Subst[Int[(d+c*x)^(p-n)*(1-x^2/a^2)^(n/2)/x^(m+p+2),x],x,1/x] /;
FreeQ[{a,c,d},x] && EqQ[a*c+d,0] && IntegerQ[(n-1)/2] && IntegerQ[m] && IntegerQ[p] *)
```

2: $\int x^m (c+d x)^p e^{n \operatorname{ArcCoth}[a x]} dx$ when $a c + d == 0 \land \frac{n-1}{2} \in \mathbb{Z} \land m \in \mathbb{Z} \land p - \frac{1}{2} \in \mathbb{Z}$

Derivation: Algebraic simplification, integration by substitution and piecewise constant extraction!

Basis: If $n \in \mathbb{Z}$, then $(c - a c x)^n e^{n \operatorname{ArcCoth}[a x]} == (-a)^n c^n x^n \left(1 - \frac{1}{a^2 x^2}\right)^{n/2}$

Basis: $F\left[\frac{1}{x}\right] == -\operatorname{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Basis: $\partial_x \frac{\sqrt{c+d x}}{\sqrt{x} \sqrt{d+\frac{c}{x}}} == 0$

Rule: If $a c + d == 0 \land \frac{n-1}{2} \in \mathbb{Z} \land m \in \mathbb{Z} \land p - \frac{1}{2} \in \mathbb{Z}$, then

$$\int x^m (c+d x)^p e^{n \operatorname{ArcCoth}[a x]} dx \longrightarrow (-a)^n c^n \int x^{m+n} (c+d x)^{p-n} \left(1 - \frac{1}{a^2 x^2}\right)^{n/2} dx$$

$$\longrightarrow \frac{(-a)^n c^n \sqrt{c+d x}}{\sqrt{x} \sqrt{d+\frac{c}{x}}} \int \frac{\left(d+\frac{c}{x}\right)^{p-n} \left(1-\frac{1}{a^2 x^2}\right)^{n/2}}{\left(\frac{1}{x}\right)^{m+p}} dx \longrightarrow -\frac{(-a)^n c^n \sqrt{c+d x}}{\sqrt{x} \sqrt{d+\frac{c}{x}}} \operatorname{Subst}\left[\int \frac{(d+c x)^{p-n} \left(1-\frac{x^2}{a^2}\right)^{n/2}}{x^{m+p+2}} dx, x, \frac{1}{x}\right]$$

Program code:

```
(* Int[(c_+d_.*x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -(-a)^n*c^n*Sqrt[c+d*x]/(Sqrt[x]*Sqrt[d+c/x])*Subst[Int[(d+c*x)^(p-n)*(1-x^2/a^2)^(n/2)/x^(p+2),x],x,1/x] /;
FreeQ[{a,c,d},x] && EqQ[a*c+d,0] && IntegerQ[(n-1)/2] && IntegerQ[p-1/2] *)
```

```
(* Int[x_^m_.*(c_+d_.*x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -(-a)^n*c^n*Sqrt[c+d*x]/(Sqrt[x]*Sqrt[d+c/x])*Subst[Int[(d+c*x)^(p-n)*(1-x^2/a^2)^(n/2)/x^(m+p+2),x],x,1/x] /;
FreeQ[{a,c,d},x] && EqQ[a*c+d,0] && IntegerQ[(n-1)/2] && IntegerQ[m] && IntegerQ[p-1/2] *)
```

1: $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $a^2\,c^2-d^2==0\ \land\ \frac{n}{2}\notin\mathbb{Z}\ \land\ p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $p\in\mathbb{Z}$, then $(c+d\,x)^p==d^p\,x^p\left(1+\frac{c}{d\,x}\right)^p$

Rule: If $a^2\,c^2-d^2==0\ \land\ \frac{n}{2}\notin\mathbb{Z}\ \land\ p\in\mathbb{Z}$, then

$$\int u\,(c+d\,x)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx\ \longrightarrow\ d^p\int u\,x^p\left(1+\frac{c}{d\,x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  d^p*Int[u*x^p*(1+c/(d*x))^p*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c^2-d^2,0] && Not[IntegerQ[n/2]] && IntegerQ[p]
```

2: $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $a^2\,c^2-d^2=0\,\land\,\frac{n}{2}\notin\mathbb{Z}\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\,\frac{(c+d\,x)^p}{x^p\,\left(1+\frac{c}{d\,x}\right)^p}=0$

Rule: If $a^2\,c^2-d^2=0\,\land\,\frac{n}{2}\notin\mathbb{Z}\,\land\,p\notin\mathbb{Z}$, then

$$\int u\,(c+d\,x)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx\,\longrightarrow\,\frac{(c+d\,x)^p}{x^p\,\left(1+\frac{c}{d\,x}\right)^p}\int u\,x^p\left(1+\frac{c}{d\,x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_)^p_*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  (c+d*x)^p/(x^p*(1+c/(d*x))^p)*Int[u*x^p*(1+c/(d*x))^p*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c^2-d^2,0] && Not[IntegerQ[n/2]] && Not[IntegerQ[p]]
```

3. $\int u \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c^2-a^2 d^2==0 \land \frac{n}{2}\notin\mathbb{Z}$

   1. $\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c^2-a^2 d^2==0 \land \frac{n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0)$

      1. $\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c^2-a^2 d^2==0 \land \frac{n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0) \land m\in\mathbb{Z}$

         1: $\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c+a\,d==0 \land \frac{n-1}{2}\in\mathbb{Z} \land m\in\mathbb{Z} \land \left(p\in\mathbb{Z} \lor p-\frac{n}{2}==0 \lor p-\frac{n}{2}-1==0\right)$

Derivation: Algebraic simplification and integration by substitution

Basis: If $c+a\,d==0 \land n\in\mathbb{Z}$, then $\left(c+\frac{d}{x}\right)^n e^{n\,\text{ArcCoth}[a\,x]} == c^n \left(1-\frac{1}{a^2 x^2}\right)^{n/2}$

Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{1}{x}\right]\,\partial_x \frac{1}{x}$

Note: The condition $p\in\mathbb{Z} \lor p-\frac{n}{2}==0 \lor p-\frac{n}{2}-1==0$ should be removed when the rules for integrands of the form $(d+e\,x)^m\,(f+g\,x)^n\,(a+b\,x+c\,x^2)^p$ when $c\,d^2-b\,d\,e+a\,e^2==0$ are strengthened.

Rule: If $c+a\,d==0 \land \frac{n-1}{2}\in\mathbb{Z} \land m\in\mathbb{Z}$, then

$$\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx \to c^n \int \frac{\left(c+\frac{d}{x}\right)^{p-n}\left(1-\frac{1}{a^2 x^2}\right)^{n/2}}{\left(\frac{1}{x}\right)^m}\,dx \to -c^n\,\text{Subst}\left[\int \frac{(c+d\,x)^{p-n}\left(1-\frac{x^2}{a^2}\right)^{n/2}}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(c_+d_./x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -c^n*Subst[Int[(c+d*x)^(p-n)*(1-x^2/a^2)^(n/2)/x^2,x],x,1/x] /;
FreeQ[{a,c,d,p},x] && EqQ[c+a*d,0] && IntegerQ[(n-1)/2] && (IntegerQ[p] || EqQ[p,n/2] || EqQ[p,n/2+1]) && IntegerQ[2*p]
```

```
Int[x_^m_.*(c_+d_./x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -c^n*Subst[Int[(c+d*x)^(p-n)*(1-x^2/a^2)^(n/2)/x^(m+2),x],x,1/x] /;
FreeQ[{a,c,d,p},x] && EqQ[c+a*d,0] && IntegerQ[(n-1)/2] && IntegerQ[m] && (IntegerQ[p] || EqQ[p,n/2] || EqQ[p,n/2+1] || LtQ[-5,m,-1]) && Inte
```

2: $\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c^2-a^2 d^2==0 \land \frac{n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0) \land m\in\mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $e^{n\,\text{ArcCoth}[z]}==\frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $F\left[\frac{1}{x}\right]==-\text{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Note: Since $c^2-a^2 d^2==0$, the factor $\left(1+\frac{d\,x}{c}\right)^p$ will combine with the factor $\left(1+\frac{x}{a}\right)^{n/2}$ or $\left(1-\frac{x}{a}\right)^{-n/2}$.

Rule: If $c^2-a^2 d^2==0 \land \frac{n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0) \land m\in\mathbb{Z}$, then

$$\int x^m\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx \longrightarrow c^p\int\frac{1}{\left(\frac{1}{x}\right)^m}\left(1+\frac{d}{c\,x}\right)^p\frac{\left(1+\frac{1}{a\,x}\right)^{n/2}}{\left(1-\frac{1}{a\,x}\right)^{n/2}}\,dx \longrightarrow -c^p\,\text{Subst}\left[\int\frac{\left(1+\frac{d\,x}{c}\right)^p\left(1+\frac{x}{a}\right)^{n/2}}{x^{m+2}\left(1-\frac{x}{a}\right)^{n/2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(c_+d_./x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -c^p*Subst[Int[(1+d*x/c)^p*(1+x/a)^(n/2)/(x^2*(1-x/a)^(n/2)),x],x,1/x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c^2-a^2*d^2,0] && Not[IntegerQ[n/2]] && (IntegerQ[p] || GtQ[c,0])
```

```
Int[x_^m_.*(c_+d_./x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -c^p*Subst[Int[(1+d*x/c)^p*(1+x/a)^(n/2)/(x^(m+2)*(1-x/a)^(n/2)),x],x,1/x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c^2-a^2*d^2,0] && Not[IntegerQ[n/2]] && (IntegerQ[p] || GtQ[c,0]) && IntegerQ[m]
```

2: $\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c^2-a^2 d^2==0 \land \frac{n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0) \land m\notin\mathbb{Z}$

Derivation: Algebraic simplification, piecewise constant extraction and integration by substitution

Basis: $e^{n\,\text{ArcCoth}[z]} = \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $\partial_x \left(x^m \left(\frac{1}{x}\right)^m\right) = 0$

Basis: $F\left[\frac{1}{x}\right] = -\text{Subst}\left[\frac{F[x]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Note: Since $c^2 - a^2 d^2 = 0$, the factor $\left(1+\frac{d\,x}{c}\right)^p$ will combine with the factor $\left(1+\frac{x}{a}\right)^{n/2}$ or $\left(1-\frac{x}{a}\right)^{-n/2}$.

Rule: If $c^2 - a^2 d^2 = 0 \land \frac{n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c > 0) \land m \notin \mathbb{Z}$, then

$$\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx \longrightarrow c^p x^m \left(\frac{1}{x}\right)^m \int \frac{1}{\left(\frac{1}{x}\right)^m} \left(1+\frac{d}{c\,x}\right)^p \frac{\left(1+\frac{1}{a\,x}\right)^{n/2}}{\left(1-\frac{1}{a\,x}\right)^{n/2}}\,dx$$

$$\longrightarrow -c^p x^m \left(\frac{1}{x}\right)^m \text{Subst}\left[\int \frac{\left(1+\frac{d\,x}{c}\right)^p \left(1+\frac{x}{a}\right)^{n/2}}{x^{m+2} \left(1-\frac{x}{a}\right)^{n/2}}\,dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_*(c_+d_./x_)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -c^p*x^m*(1/x)^m*Subst[Int[(1+d*x/c)^p*(1+x/a)^(n/2)/(x^(m+2)*(1-x/a)^(n/2)),x],x,1/x] /;
FreeQ[{a,c,d,m,n,p},x] && EqQ[c^2-a^2*d^2,0] && Not[IntegerQ[n/2]] && (IntegerQ[p] || GtQ[c,0]) && Not[IntegerQ[m]]
```

2: $\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx$ when $c^2 - a^2 d^2 == 0 \land \frac{n}{2} \notin \mathbb{Z} \land \neg (p \in \mathbb{Z} \lor c > 0)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(c+\frac{d}{x}\right)^p}{\left(1+\frac{d}{c x}\right)^p} == 0$

Rule: If $c^2 - a^2 d^2 == 0 \land \frac{n}{2} \notin \mathbb{Z} \land \neg (p \in \mathbb{Z} \lor c > 0)$, then

$$\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx \longrightarrow \frac{\left(c + \frac{d}{x}\right)^p}{\left(1 + \frac{d}{c x}\right)^p} \int u \left(1 + \frac{d}{c x}\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx$$

Program code:

```
Int[u_.*(c_+d_./x_)^p_*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  (c+d/x)^p/(1+d/(c*x))^p*Int[u*(1+d/(c*x))^p*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c^2-a^2*d^2,0] && Not[IntegerQ[n/2]] && Not[IntegerQ[p] || GtQ[c,0]]
```

4. $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $a^2\,c+d==0\,\land\,\frac{n}{2}\notin\mathbb{Z}$

1. $\int (c+d\,x^2)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $a^2\,c+d==0\,\land\,\frac{n}{2}\notin\mathbb{Z}\,\land\,p\le -1$

1: $\int \frac{e^{n\,\text{ArcCoth}[a\,x]}}{c+d\,x^2}\,dx$ when $a^2\,c+d==0\,\land\,\frac{n}{2}\notin\mathbb{Z}$

Rule: If $a^2\,c+d==0\,\land\,\frac{n}{2}\notin\mathbb{Z}$, then

$$\int \frac{e^{n\,\text{ArcCoth}[a\,x]}}{c+d\,x^2}\,dx\,\longrightarrow\,\frac{e^{n\,\text{ArcCoth}[a\,x]}}{a\,c\,n}$$

Program code:

```
Int[E^(n_.*ArcCoth[a_.*x_])/(c_+d_.*x_^2),x_Symbol] :=
  E^(n*ArcCoth[a*x])/(a*c*n) /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n/2]]
```

2: $\int \frac{e^{n\,\text{ArcCoth}[a\,x]}}{(c+d\,x^2)^{3/2}}\,dx$ when $a^2\,c+d==0\,\land\,n\notin\mathbb{Z}$

Note: When n is an integer, it is better to transform integrand into algebraic form.

Rule: If $a^2\,c+d==0\,\land\,n\notin\mathbb{Z}$, then

$$\int \frac{e^{n\,\text{ArcCoth}[a\,x]}}{(c+d\,x^2)^{3/2}}\,dx\,\longrightarrow\,\frac{(n-a\,x)\,e^{n\,\text{ArcCoth}[a\,x]}}{a\,c\,(n^2-1)\,\sqrt{c+d\,x^2}}$$

Program code:

```
Int[E^(n_*ArcCoth[a_.*x_])/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  (n-a*x)*E^(n*ArcCoth[a*x])/(a*c*(n^2-1)*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n]]
```

3: $\int (c+d x^2)^p e^{n \text{ArcCoth}[a x]} dx$ when $a^2 c+d=0 \land \frac{n}{2} \notin \mathbb{Z} \land p<-1 \land p \neq -\frac{3}{2} \land n^2-4 (p+1)^2 \neq 0$

- Rule: If $a^2 c+d=0 \land \frac{n}{2} \notin \mathbb{Z} \land p<-1 \land p \neq -\frac{3}{2} \land n^2-4 (p+1)^2 \neq 0$, then

$$\int (c+d x^2)^p e^{n \text{ArcCoth}[a x]} dx \to \frac{(n+2 a (p+1) x) (c+d x^2)^{p+1} e^{n \text{ArcCoth}[a x]}}{a c (n^2-4 (p+1)^2)} - \frac{2 (p+1) (2 p+3)}{c (n^2-4 (p+1)^2)} \int (c+d x^2)^{p+1} e^{n \text{ArcCoth}[a x]} dx$$

- Program code:

```
Int[(c_+d_.*x_^2)^p_*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  (n+2*a*(p+1)*x)*(c+d*x^2)^(p+1)*E^(n*ArcCoth[a*x])/(a*c*(n^2-4*(p+1)^2)) -
  2*(p+1)*(2*p+3)/(c*(n^2-4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n/2]] && LtQ[p,-1] && NeQ[p,-3/2] && NeQ[n^2-4*(p+1)^2,0] && (IntegerQ[p] || Not[Integer
```

2. $\int x^m (c+d x^2)^p e^{n \text{ArcCoth}[a x]} dx$ when $a^2 c+d=0 \land \frac{n}{2} \notin \mathbb{Z} \land m \in \mathbb{Z} \land 0 \leq m \leq -2 (p+1)$

1. $\int x (c+d x^2)^p e^{n \text{ArcCoth}[a x]} dx$ when $a^2 c+d=0 \land \frac{n}{2} \notin \mathbb{Z} \land p \leq -1$

1: $\int \frac{x e^{n \text{ArcCoth}[a x]}}{(c+d x^2)^{3/2}} dx$ when $a^2 c+d=0 \land n \notin \mathbb{Z}$

- Rule: If $a^2 c+d=0 \land n \notin \mathbb{Z}$, then

$$\int \frac{x e^{n \text{ArcCoth}[a x]}}{(c+d x^2)^{3/2}} dx \to -\frac{(1-a n x) e^{n \text{ArcCoth}[a x]}}{a^2 c (n^2-1) \sqrt{c+d x^2}}$$

- Program code:

```
Int[x_*E^(n_*ArcCoth[a_.*x_])/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  -(1-a*n*x)*E^(n*ArcCoth[a*x])/(a^2*c*(n^2-1)*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n]]
```

2: $\int x\left(c+d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx$ when $a^2 c+d=0 \wedge \frac{n}{2} \notin \mathbb{Z} \wedge p \leq -1 \wedge p \neq -\frac{3}{2} \wedge n^2-4(p+1)^2 \neq 0$

Rule: If $a^2 c+d=0 \wedge \frac{n}{2} \notin \mathbb{Z} \wedge p \leq -1 \wedge p \neq -\frac{3}{2} \wedge n^2-4(p+1)^2 \neq 0 \wedge p \notin \mathbb{Z}$, then

$$\int x\left(c+d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx \longrightarrow \frac{(2(p+1)+a n x)\left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcCoth}[a x]}}{a^2 c\left(n^2-4(p+1)^2\right)}-\frac{n(2 p+3)}{a c\left(n^2-4(p+1)^2\right)} \int\left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcCoth}[a x]} \, dx$$

Program code:

```
Int[x_*(c_+d_.*x_^2)^p_*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  (2*(p+1)+a*n*x)*(c+d*x^2)^(p+1)*E^(n*ArcCoth[a*x])/(a^2*c*(n^2-4*(p+1)^2)) -
  n*(2*p+3)/(a*c*(n^2-4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n/2]] && LeQ[p,-1] && NeQ[p,-3/2] && NeQ[n^2-4*(p+1)^2,0] && (IntegerQ[p] || Not[Integer
```

2. $\int x^2\left(c+d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx$ when $a^2 c+d=0 \wedge \frac{n}{2} \notin \mathbb{Z} \wedge p \leq -1$

1: $\int x^2\left(c+d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx$ when $a^2 c+d=0 \wedge \frac{n}{2} \notin \mathbb{Z} \wedge n^2+2(p+1)=0 \wedge n^2 \neq 1$

Rule: If $a^2 c+d=0 \wedge \frac{n}{2} \notin \mathbb{Z} \wedge n^2+2(p+1)=0 \wedge n^2 \neq 1$, then

$$\int x^2\left(c+d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx \longrightarrow -\frac{(n+2(p+1) a x)\left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcCoth}[a x]}}{a^3 c n^2\left(n^2-1\right)}$$

Program code:

```
Int[x_^2*(c_+d_.*x_^2)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -(n+2*(p+1)*a*x)*(c+d*x^2)^(p+1)*E^(n*ArcCoth[a*x])/(a^3*c*n^2*(n^2-1)) /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n/2]] && EqQ[n^2+2*(p+1),0] && NeQ[n^2,1]
```

2: $\int x^2 \left(c + d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx$ when $a^2 c + d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land p \le -1 \land n^2 + 2 (p + 1) \neq 0 \land n^2 - 4 (p + 1)^2 \neq 0$

Rule: If $a^2 c + d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land p \le -1 \land n^2 + 2 (p + 1) \neq 0 \land n^2 - 4 (p + 1)^2 \neq 0$, then

$$\int x^2 \left(c + d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx \to \frac{(n + 2 (p + 1) a x) \left(c + d x^2\right)^{p+1} e^{n \operatorname{ArcCoth}[a x]}}{a^3 c \left(n^2 - 4 (p + 1)^2\right)} - \frac{n^2 + 2 (p + 1)}{a^2 c \left(n^2 - 4 (p + 1)^2\right)} \int \left(c + d x^2\right)^{p+1} e^{n \operatorname{ArcCoth}[a x]} \, dx$$

Program code:

```
Int[x_^2*(c_+d_.*x_^2)^p_*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  (n+2*(p+1)*a*x)*(c+d*x^2)^(p+1)*E^(n*ArcCoth[a*x])/(a^3*c*(n^2-4*(p+1)^2)) -
  (n^2+2*(p+1))/(a^2*c*(n^2-4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n/2]] && LeQ[p,-1] && NeQ[n^2+2*(p+1),0] && NeQ[n^2-4*(p+1)^2,0] &&
  (IntegerQ[p] || Not[IntegerQ[n]])
```

3: $\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx$ when $a^2 c + d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land m \in \mathbb{Z} \land 3 \le m \le -2 (p + 1) \land p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $a^2 c + d == 0 \land m \in \mathbb{Z} \land p \in \mathbb{Z}$, then
$x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} == -\frac{(-c)^p}{a^{m+1}} \frac{e^{n \operatorname{ArcCoth}[a x]} \operatorname{Coth}[\operatorname{ArcCoth}[a x]]^{m+2 (p+1)}}{\operatorname{Cosh}[\operatorname{ArcCoth}[a x]]^{2 (p+1)}} \partial_x \operatorname{ArcCoth}[a x]$

Rule: If $a^2 c + d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land m \in \mathbb{Z} \land 3 \le m \le -2 (p + 1) \land p \in \mathbb{Z}$, then

$$\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcCoth}[a x]} \, dx \to -\frac{(-c)^p}{a^{m+1}} \operatorname{Subst}\left[\int \frac{e^{n x} \operatorname{Coth}[x]^{m+2 (p+1)}}{\operatorname{Cosh}[x]^{2 (p+1)}} \, dx, x, \operatorname{ArcCoth}[a x]\right]$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -(-c)^p/a^(m+1)*Subst[Int[E^(n*x)*Coth[x]^(m+2*(p+1))/Cosh[x]^(2*(p+1)),x],x,ArcCoth[a*x]] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n/2]] && IntegerQ[m] && LeQ[3,m,-2(p+1)] && IntegerQ[p]
```

3. $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $a^2\,c+d==0\ \land\ \frac{n}{2}\notin\mathbb{Z}$

1: $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $a^2\,c+d==0\ \land\ \frac{n}{2}\notin\mathbb{Z}\ \land\ p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $a^2\,c+d==0\ \land\ p\in\mathbb{Z}$, then $(c+d\,x^2)^p==d^p\,x^{2\,p}\left(1-\frac{1}{a^2\,x^2}\right)^p$

Rule: If $a^2\,c+d==0\ \land\ \frac{n}{2}\notin\mathbb{Z}\ \land\ p\in\mathbb{Z}$, then

$$\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcCoth}[a\,x]}\,dx\ \longrightarrow\ d^p\int u\,x^{2\,p}\left(1-\frac{1}{a^2\,x^2}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_^2)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  d^p*Int[u*x^(2*p)*(1-1/(a^2*x^2))^p*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n/2]] && IntegerQ[p]
```

2: $\int u\left(c+d\,x^2\right)^p e^{n\,\mathrm{ArcCoth}[a\,x]}\,dx$ when $a^2\,c+d=0\,\land\,\frac{n}{2}\notin\mathbb{Z}\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $a^2\,c+d=0$, then $\partial_x \frac{\left(c+d\,x^2\right)^p}{x^{2\,p}\left(1-\frac{1}{a^2\,x^2}\right)^p}=0$

Rule: If $a^2\,c+d=0\,\land\,\frac{n}{2}\notin\mathbb{Z}\,\land\,p\notin\mathbb{Z}$, then

$$\int u\left(c+d\,x^2\right)^p e^{n\,\mathrm{ArcCoth}[a\,x]}\,dx\,\longrightarrow\,\frac{\left(c+d\,x^2\right)^p}{x^{2\,p}\left(1-\frac{1}{a^2\,x^2}\right)^p}\int u\,x^{2\,p}\left(1-\frac{1}{a^2\,x^2}\right)^p e^{n\,\mathrm{ArcCoth}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_^2)^p_*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  (c+d*x^2)^p/(x^(2*p)*(1-1/(a^2*x^2))^p)*Int[u*x^(2*p)*(1-1/(a^2*x^2))^p*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c+d,0] && Not[IntegerQ[n/2]] && Not[IntegerQ[p]]
```

5. $\int u \left(c+\frac{d}{x^2}\right)^p e^{n \text{ArcCoth}[a x]} \, dx$ when $c+a^2 d == 0 \land \frac{n}{2} \notin \mathbb{Z}$

1. $\int u \left(c+\frac{d}{x^2}\right)^p e^{n \text{ArcCoth}[a x]} \, dx$ when $c+a^2 d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c>0)$

1: $\int u \left(c+\frac{d}{x^2}\right)^p e^{n \text{ArcCoth}[a x]} \, dx$ when $c+a^2 d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c>0) \land \left(2 p \mid p+\frac{n}{2}\right) \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n \text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $\left(1-z^2\right)^p == (1-z)^p (1+z)^p$

Basis: If $p+n \in \mathbb{Z}$, then $\left(1-\frac{1}{z}\right)^{p-n} \left(1+\frac{1}{z}\right)^{p+n} == \frac{(-1+z)^{p-n} (1+z)^{p+n}}{z^{2 p}}$

Rule: If $c+a^2 d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c>0) \land \left(2 p \mid p+\frac{n}{2}\right) \in \mathbb{Z}$, then

$$\int u \left(c+\frac{d}{x^2}\right)^p e^{n \text{ArcCoth}[a x]} \, dx \to c^p \int u \left(1-\frac{1}{a^2 x^2}\right)^p \frac{\left(1+\frac{1}{a x}\right)^{n/2}}{\left(1-\frac{1}{a x}\right)^{n/2}} \, dx$$

$$\to c^p \int u \left(1-\frac{1}{a x}\right)^{p-\frac{n}{2}} \left(1+\frac{1}{a x}\right)^{p+\frac{n}{2}} \, dx$$

$$\to \frac{c^p}{a^{2 p}} \int \frac{u}{x^{2 p}} (-1+a x)^{p-\frac{n}{2}} (1+a x)^{p+\frac{n}{2}} \, dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  c^p/a^(2*p)*Int[u/x^(2*p)*(-1+a*x)^(p-n/2)*(1+a*x)^(p+n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c+a^2*d,0] && Not[IntegerQ[n/2]] && (IntegerQ[p] || GtQ[c,0]) && IntegersQ[2*p,p+n/2]
```

2. $\int x^m \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c+a^2 d=0 \land \frac{n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0) \land \neg\left(2p \mid p+\frac{n}{2}\right)\in\mathbb{Z}$

1: $\int x^m \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c+a^2 d=0 \land \frac{n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0) \land \neg\left(2p \mid p+\frac{n}{2}\right)\in\mathbb{Z} \land m\in\mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $e^{n\,\text{ArcCoth}[z]} = \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $\left(1-z^2\right)^p = (1-z)^p\,(1+z)^p$

Basis: $F\left[\frac{1}{x}\right] = -\text{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Rule: If $c+a^2 d=0 \land \frac{n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0) \land \neg\left(2p \mid p+\frac{n}{2}\right)\in\mathbb{Z}$, then

$$\int x^m \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx \to c^p\int x^m\left(1-\frac{1}{a^2x^2}\right)^p\frac{\left(1+\frac{1}{a\,x}\right)^{n/2}}{\left(1-\frac{1}{a\,x}\right)^{n/2}}\,dx$$

$$\to c^p\int\frac{1}{\left(\frac{1}{x}\right)^m}\left(1-\frac{1}{a\,x}\right)^{p-\frac{n}{2}}\left(1+\frac{1}{a\,x}\right)^{p+\frac{n}{2}}\,dx$$

$$\to -c^p\,\text{Subst}\left[\int\frac{\left(1-\frac{x}{a}\right)^{p-\frac{n}{2}}\left(1+\frac{x}{a}\right)^{p+\frac{n}{2}}}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(c_+d_./x_^2)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -c^p*Subst[Int[(1-x/a)^(p-n/2)*(1+x/a)^(p+n/2)/x^2,x],x,1/x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c+a^2*d,0] && Not[IntegerQ[n/2]] && (IntegerQ[p] || GtQ[c,0]) && Not[IntegersQ[2*p,p+n/2]]
```

```
Int[x_^m_.*(c_+d_./x_^2)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -c^p*Subst[Int[(1-x/a)^(p-n/2)*(1+x/a)^(p+n/2)/x^(m+2),x],x,1/x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c+a^2*d,0] && Not[IntegerQ[n/2]] && (IntegerQ[p] || GtQ[c,0]) && Not[IntegersQ[2*p,p+n/2]] && IntegerQ[m]
```

2: $\int x^m \left(c + \frac{d}{x^2}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c + a^2 d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c > 0) \land \neg \left(2p \mid p + \frac{n}{2}\right) \in \mathbb{Z} \land m \notin \mathbb{Z}$

Derivation: Algebraic simplification, piecewise constant extraction and integration by substitution

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $\left(1 - z^2\right)^p == (1 - z)^p\,(1 + z)^p$

Basis: $\partial_x \left(x^m \left(\frac{1}{x}\right)^m\right) == 0$

Basis: $F\left[\frac{1}{x}\right] == -\text{Subst}\left[\frac{F[x]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x \frac{1}{x}$

Rule: If $c + a^2 d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c > 0) \land \neg \left(2p \mid p + \frac{n}{2}\right) \in \mathbb{Z}$, then

$$\int x^m \left(c + \frac{d}{x^2}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx \longrightarrow c^p \int x^m \left(1 - \frac{1}{a^2 x^2}\right)^p \frac{\left(1 + \frac{1}{a\,x}\right)^{n/2}}{\left(1 - \frac{1}{a\,x}\right)^{n/2}}\,dx$$

$$\longrightarrow c^p x^m \left(\frac{1}{x}\right)^m \int \frac{1}{\left(\frac{1}{x}\right)^m} \left(1 - \frac{1}{a\,x}\right)^{p-\frac{n}{2}} \left(1 + \frac{1}{a\,x}\right)^{p+\frac{n}{2}}\,dx$$

$$\longrightarrow -c^p x^m \left(\frac{1}{x}\right)^m \text{Subst}\left[\int \frac{\left(1 - \frac{x}{a}\right)^{p-\frac{n}{2}} \left(1 + \frac{x}{a}\right)^{p+\frac{n}{2}}}{x^{m+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[x_^m_*(c_+d_./x_^2)^p_.*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  -c^p*x^m*(1/x)^m*Subst[Int[(1-x/a)^(p-n/2)*(1+x/a)^(p+n/2)/x^(m+2),x],x,1/x] /;
FreeQ[{a,c,d,m,n,p},x] && EqQ[c+a^2*d,0] && Not[IntegerQ[n/2]] && (IntegerQ[p] || GtQ[c,0]) && Not[IntegersQ[2*p,p+n/2]] && Not[IntegerQ[m]]
```

2: $\int u \left(c + \frac{d}{x^2}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$ when $c + a^2 d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land \neg (p \in \mathbb{Z} \lor c > 0)$

Derivation: Piecewise constant extraction

Basis: If $c + a^2 d == 0$, then $\partial_x \frac{\left(c+\frac{d}{x^2}\right)^p}{\left(1-\frac{1}{a^2 x^2}\right)^p} == 0$

Rule: If $c + a^2 d == 0 \land \frac{n}{2} \notin \mathbb{Z} \land \neg (p \in \mathbb{Z} \lor c > 0)$, then

$$\int u \left(c + \frac{d}{x^2}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx \to \frac{c^{\text{IntPart}[p]} \left(c + \frac{d}{x^2}\right)^{\text{FracPart}[p]}}{\left(1 - \frac{1}{a^2 x^2}\right)^{\text{FracPart}[p]}} \int u \left(1 - \frac{1}{a^2 x^2}\right)^p e^{n\,\text{ArcCoth}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_*E^(n_.*ArcCoth[a_.*x_]),x_Symbol] :=
  c^IntPart[p]*(c+d/x^2)^FracPart[p]/(1-1/(a^2*x^2))^FracPart[p]*Int[u*(1-1/(a^2*x^2))^p*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c+a^2*d,0] && Not[IntegerQ[n/2]] && Not[IntegerQ[p] || GtQ[c,0]]
```

2. $\int u\, e^{n\,\text{ArcCoth}[a+b\,x]}\,dx$

1: $\int u\, e^{n\,\text{ArcCoth}[a+b\,x]}\,dx$ when $\frac{n}{2}\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $\frac{n}{2}\in\mathbb{Z}$, then $e^{n\,\text{ArcCoth}[z]} == (-1)^{n/2}\, e^{n\,\text{ArcTanh}[z]}$

Rule: If $\frac{n}{2}\in\mathbb{Z}$, then

$$\int u\, e^{n\,\text{ArcCoth}[c\,(a+b\,x)]}\,dx \longrightarrow (-1)^{n/2}\int u\, e^{n\,\text{ArcTanh}[c\,(a+b\,x)]}\,dx$$

Program code:

```
Int[u_.*E^(n_*ArcCoth[c_.*(a_+b_.*x_)]),x_Symbol] :=
  (-1)^(n/2)*Int[u*E^(n*ArcTanh[c*(a+b*x)]),x] /;
FreeQ[{a,b,c},x] && IntegerQ[n/2]
```

2. $\int u\, e^{n\,\text{ArcCoth}[a+b\,x]}\,dx$ when $\frac{n}{2}\notin\mathbb{Z}$

1: $\int e^{n\,\text{ArcCoth}[c\,(a+b\,x)]}\,dx$ when $\frac{n}{2}\notin\mathbb{Z}$

Derivation: Algebraic simplification and piecewise constant extraction

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}} == \frac{z^{n/2}\left(1+\frac{1}{z}\right)^{n/2}}{(-1+z)^{n/2}}$

Basis: $\partial_x \frac{f[x]^n\left(1+\frac{1}{f[x]}\right)^n}{(1+f[x])^n} == 0$

Rule: If $\frac{n}{2}\notin\mathbb{Z}$, then

$$\int e^{n\,\text{ArcCoth}[c\,(a+b\,x)]}\,dx \longrightarrow \int \frac{(c\,(a+b\,x))^{n/2}\left(1+\frac{1}{c\,(a+b\,x)}\right)^{n/2}}{(-1+c\,(a+b\,x))^{n/2}}\,dx \longrightarrow \frac{(c\,(a+b\,x))^{n/2}\left(1+\frac{1}{c\,(a+b\,x)}\right)^{n/2}}{(1+a\,c+b\,c\,x)^{n/2}}\int \frac{(1+a\,c+b\,c\,x)^{n/2}}{(-1+a\,c+b\,c\,x)^{n/2}}\,dx$$

–

Program code:

```
Int[E^(n_.*ArcCoth[c_.*(a_+b_.*x_)]),x_Symbol] :=
  (c*(a+b*x))^(n/2)*(1+1/(c*(a+b*x)))^(n/2)/(1+a*c+b*c*x)^(n/2)*Int[(1+a*c+b*c*x)^(n/2)/(-1+a*c+b*c*x)^(n/2),x] /;
FreeQ[{a,b,c,n},x] && Not[IntegerQ[n/2]]
```

2. $\int (d+e\,x)^m\, e^{n\,\text{ArcCoth}[c\,(a+b\,x)]}\, dx$ when $\frac{n}{2} \notin \mathbb{Z}$

1: $\int x^m\, e^{n\,\text{ArcCoth}[c\,(a+b\,x)]}\, dx$ when $m \in \mathbb{Z}^- \land -1 < n < 1$

Derivation: Algebraic simplification and integration by substitution

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: If $m \in \mathbb{Z} \land -1 < n < 1$, then
$x^m \frac{\left(1+\frac{1}{c\,(a+b\,x)}\right)^{n/2}}{\left(1-\frac{1}{c\,(a+b\,x)}\right)^{n/2}} == -\frac{4}{n\,b^{m+1}\,c^{m+1}}\,\text{Subst}\left[\frac{x^{2/n}\left(1+a\,c+(1-a\,c)\,x^{2/n}\right)^m}{\left(-1+x^{2/n}\right)^{m+2}},\, x,\, \frac{\left(1+\frac{1}{c\,(a+b\,x)}\right)^{n/2}}{\left(1-\frac{1}{c\,(a+b\,x)}\right)^{n/2}}\right]\,\partial_x \frac{\left(1+\frac{1}{c\,(a+b\,x)}\right)^{n/2}}{\left(1-\frac{1}{c\,(a+b\,x)}\right)^{n/2}}$

Note: There should be an algebraic substitution rule that makes this rule redundant.

Rule: If $m \in \mathbb{Z}^- \land -1 < n < 1$, then

$$\int x^m\, e^{n\,\text{ArcCoth}[c\,(a+b\,x)]}\, dx \to \int x^m \frac{\left(1+\frac{1}{c\,(a+b\,x)}\right)^{n/2}}{\left(1-\frac{1}{c\,(a+b\,x)}\right)^{n/2}}\, dx$$

$$\to -\frac{4}{n\,b^{m+1}\,c^{m+1}}\,\text{Subst}\left[\int \frac{x^{2/n}\left(1+a\,c+(1-a\,c)\,x^{2/n}\right)^m}{\left(-1+x^{2/n}\right)^{m+2}}\, dx,\, x,\, \frac{\left(1+\frac{1}{c\,(a+b\,x)}\right)^{n/2}}{\left(1-\frac{1}{c\,(a+b\,x)}\right)^{n/2}}\right]$$

Program code:

```
Int[x_^m_*E^(n_*ArcCoth[c_.*(a_+b_.*x_)]),x_Symbol] :=
  -4/(n*b^(m+1)*c^(m+1))*
    Subst[Int[x^(2/n)*(1+a*c+(1-a*c)*x^(2/n))^m/(-1+x^(2/n))^(m+2),x],x,(1+1/(c*(a+b*x)))^(n/2)/(1-1/(c*(a+b*x)))^(n/2)] /;
FreeQ[{a,b,c},x] && ILtQ[m,0] && LtQ[-1,n,1]
```

2: $\int (d+e\,x)^m\, e^{n\,\text{ArcCoth}[c\,(a+b\,x)]}\, dx$ when $\frac{n}{2}\notin\mathbb{Z}$

Derivation: Algebraic simplification and piecewise constant extraction

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}} == \frac{z^{n/2}\left(1+\frac{1}{z}\right)^{n/2}}{(-1+z)^{n/2}}$

Basis: $\partial_x \frac{f[x]^n\left(1+\frac{1}{f[x]}\right)^n}{(1+f[x])^n} == 0$

Rule: If $\frac{n}{2}\notin\mathbb{Z}$, then

$$\int (d+e\,x)^m\, e^{n\,\text{ArcCoth}[c\,(a+b\,x)]}\, dx \to \int (d+e\,x)^m\, \frac{(c\,(a+b\,x))^{n/2}\left(1+\frac{1}{c\,(a+b\,x)}\right)^{n/2}}{(-1+c\,(a+b\,x))^{n/2}}\, dx$$

$$\to \frac{(c\,(a+b\,x))^{n/2}\left(1+\frac{1}{c\,(a+b\,x)}\right)^{n/2}}{(1+a\,c+b\,c\,x)^{n/2}}\int (d+e\,x)^m\, \frac{(1+a\,c+b\,c\,x)^{n/2}}{(-1+a\,c+b\,c\,x)^{n/2}}\, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*E^(n_.*ArcCoth[c_.*(a_+b_.*x_)]),x_Symbol] :=
  (c*(a+b*x))^(n/2)*(1+1/(c*(a+b*x)))^(n/2)/(1+a*c+b*c*x)^(n/2)*Int[(d+e*x)^m*(1+a*c+b*c*x)^(n/2)/(-1+a*c+b*c*x)^(n/2),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && Not[IntegerQ[n/2]]
```

3. $\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcCoth}[a+b\,x]}\, dx$ when $\frac{n}{2}\notin\mathbb{Z}\wedge b\,d==2\,a\,e\wedge b^2\,c+e\left(1-a^2\right)==0$

1: $\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcCoth}[a+b\,x]}\, dx$ when $\frac{n}{2}\notin\mathbb{Z}\wedge b\,d==2\,a\,e\wedge b^2\,c+e\left(1-a^2\right)==0\wedge\left(p\in\mathbb{Z}\vee\frac{c}{1-a^2}>0\right)$

Derivation: Algebraic simplification and piecewise constant extraction

Basis: If $b\,d==2\,a\,e\wedge b^2\,c+e\left(1-a^2\right)==0$, then $c+d\,x+e\,x^2==\frac{c}{1-a^2}\left(1-(a+b\,x)^2\right)$

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}} == \frac{z^{n/2}\left(1+\frac{1}{z}\right)^{n/2}}{(-1+z)^{n/2}}$

Basis: $\partial_x \frac{(a+b\,x)^n\left(1+\frac{1}{a+b\,x}\right)^n}{(1+a+b\,x)^n} == 0$

Basis: $\partial_x \frac{(1-a-b\,x)^n}{(-1+a+b\,x)^n} == 0$

Basis: $\left(1-z^2\right)^p == (1-z)^p\,(1+z)^p$

Basis: $\frac{z^n\left(1+\frac{1}{z}\right)^n}{(1+z)^n} == \left(\frac{z}{1+z}\right)^n\left(\frac{1+z}{z}\right)^n$

Rule: If $\frac{n}{2} \notin \mathbb{Z} \land b\,d == 2\,a\,e \land b^2\,c + e\left(1-a^2\right) == 0 \land \left(p \in \mathbb{Z} \lor \frac{c}{1-a^2} > 0\right)$, then

$$\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcCoth}[a+b\,x]}\,dx \longrightarrow \left(\frac{c}{1-a^2}\right)^p \int u\left(1-(a+b\,x)^2\right)^p \frac{(a+b\,x)^{n/2}\left(1+\frac{1}{a+b\,x}\right)^{n/2}}{(-1+a+b\,x)^{n/2}}\,dx$$

$$\longrightarrow \left(\frac{c}{1-a^2}\right)^p \frac{(a+b\,x)^{n/2}\left(1+\frac{1}{a+b\,x}\right)^{n/2}}{(1+a+b\,x)^{n/2}} \frac{(1-a-b\,x)^{n/2}}{(-1+a+b\,x)^{n/2}} \int u\left(1-(a+b\,x)^2\right)^p \frac{(1+a+b\,x)^{n/2}}{(1-a-b\,x)^{n/2}}\,dx$$

$$\longrightarrow \left(\frac{c}{1-a^2}\right)^p \left(\frac{a+b\,x}{1+a+b\,x}\right)^{n/2} \left(\frac{1+a+b\,x}{a+b\,x}\right)^{n/2} \frac{(1-a-b\,x)^{n/2}}{(-1+a+b\,x)^{n/2}} \int u\,(1-a-b\,x)^{p-n/2}\,(1+a+b\,x)^{p+n/2}\,dx$$

–

Program code:

```
Int[u_.*(c_+d_.*x_+e_.*x_^2)^p_.*E^(n_.*ArcCoth[a_+b_.*x_]),x_Symbol] :=
  (c/(1-a^2))^p*((a+b*x)/(1+a+b*x))^(n/2)*((1+a+b*x)/(a+b*x))^(n/2)*((1-a-b*x)^(n/2)/(-1+a+b*x)^(n/2))*
    Int[u*(1-a-b*x)^(p-n/2)*(1+a+b*x)^(p+n/2),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && Not[IntegerQ[n/2]] && EqQ[b*d-2*a*e,0] && EqQ[b^2*c+e(1-a^2),0] && (IntegerQ[p] || GtQ[c/(1-a^2),0])
```

2: $\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcCoth}[a+b\,x]}\,dx$ when $\frac{n}{2}\notin\mathbb{Z}\,\land\,b\,d==2\,a\,e\,\land\,b^2\,c+e\left(1-a^2\right)==0\,\land\,\neg\left(p\in\mathbb{Z}\,\lor\,\frac{c}{1-a^2}>0\right)$

Derivation: Piecewise constant extraction

Basis: If $b\,d==2\,a\,e\,\land\,b^2\,c+e\left(1-a^2\right)==0$, then $\partial_x\frac{\left(c+d\,x+e\,x^2\right)^p}{\left(1-a^2-2\,a\,b\,x-b^2\,x^2\right)^p}==0$

Rule: If $b\,d==2\,a\,e\,\land\,b^2\,c+e\left(1-a^2\right)==0\,\land\,\neg\left(p\in\mathbb{Z}\,\lor\,\frac{c}{1-a^2}>0\right)$, then

$$\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcCoth}[a+b\,x]}\,dx\,\longrightarrow\,\frac{\left(c+d\,x+e\,x^2\right)^p}{\left(1-a^2-2\,a\,b\,x-b^2\,x^2\right)^p}\int u\left(1-a^2-2\,a\,b\,x-b^2\,x^2\right)^p e^{n\,\text{ArcCoth}[a+b\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_+e_.*x_^2)^p_.*E^(n_.*ArcCoth[a_+b_.*x_]),x_Symbol] :=
  (c+d*x+e*x^2)^p/(1-a^2-2*a*b*x-b^2*x^2)^p*Int[u*(1-a^2-2*a*b*x-b^2*x^2)^p*E^(n*ArcCoth[a*x]),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && Not[IntegerQ[n/2]] && EqQ[b*d-2*a*e,0] && EqQ[b^2*c+e(1-a^2),0] && Not[IntegerQ[p] || GtQ[c/(1-a^2),0]]
```

3: $\int u\, e^{n\,\text{ArcCoth}\left[\frac{c}{a+b\,x}\right]}\, dx$

Derivation: Algebraic simplification

Basis: $\text{ArcCoth}[z] == \text{ArcTanh}\left[\frac{1}{z}\right]$

Rule:

$$\int u\, e^{n\,\text{ArcCoth}\left[\frac{c}{a+b\,x}\right]}\, dx \longrightarrow \int u\, e^{n\,\text{ArcTanh}\left[\frac{a}{c}+\frac{b\,x}{c}\right]}\, dx$$

Program code:

```
Int[u_.*E^(n_.*ArcCoth[c_./(a_.+b_.*x_)]),x_Symbol] :=
  Int[u*E^(n*ArcTanh[a/c+b*x/c]),x] /;
FreeQ[{a,b,c,n},x]
```